



A MARSHALL CAVENDISH

4

COMPUTER COURSE IN WEEKLY PARTS

INFLUENT

LEARN PROGRAMMING - FOR FUN AND THE FUTURE


UK £1.00

Republic of Ireland £1.25

Malta 85c

Australia \$2.25

New Zealand \$2.95



INPUT

Vol 1

No 4

GAMES PROGRAMMING 4

SCORING AND TIMING ROUTINES 97

See how fast—and how good—you are.
Plus a simple new game program

BASIC PROGRAMMING 7

ALL ABOUT READ AND DATA 104

How to give your computer a store of
information, or operating instructions

MACHINE CODE 5

LEARNING TO COUNT ON ONE FINGER 110

Get to know binary arithmetic—the
basis of the way computers count

BASIC PROGRAMMING 8

SPOT-ON SCREEN DISPLAYS 117

Using punctuation and formatting commands for
neater layout of printing on the screen

APPLICATIONS 3

MULTIPLE LETTERS MADE EASY 124

A text editing program to speed up repetitive
letter writing—like job applications

INDEX

The last part of INPUT, Part 52, will contain a complete, cross-referenced index.
For easy access to your growing collection, a cumulative index to the contents
of each issue is contained on the inside back cover.

PICTURE CREDITS


Front Cover, Dave King. Pages 129, 130, 134, Peter Bentley. Pages 136, 138,
140, Graeme Harris. Page 142, Graeme Harris, David Lloyd. Pages 144, 146,
148, 150, Ian Craig. Pages 152, 154, Dick Ward. Page 156, Dave King. Pages
158, 160, Dave King.

© Marshall Cavendish Limited 1984/5/6

All worldwide rights reserved.

The contents of this publication including software, codes, listings,
graphics, illustrations and text are the exclusive property and copyright of
Marshall Cavendish Limited and may not be copied, reproduced,
transmitted, hired, lent, distributed, stored or modified in any form
whatsoever without the prior approval of the Copyright holder.

Published by Marshall Cavendish Partworks Ltd, 58 Old Compton Street, London W1V 5PA,
England. Printed by Artisan Press, Leicester and Howard Hunt Litho, London.



HOW TO ORDER YOUR BINDERS

UK and Republic of Ireland:
Send £4.95 (inc p & p) (IR£5.45) for
each binder to the address below:

Marshall Cavendish Services Ltd,
Department 980, Newtown Road,
Hove, Sussex BN3 7DN

Australia: See inserts for details, or
write to INPUT, Gordon and Gotch
Ltd, PO Box 213, Alexandria, NSW
2015

New Zealand: See inserts for details,
or write to INPUT, Gordon and Gotch
(NZ) Ltd, PO Box 1595, Wellington

Malta: Binders are available from
local newsagents.

BACK NUMBERS

Copies of any part of INPUT can be obtained from the following addresses at the
regular cover price, with no extra charge for postage and packing:

UK and Republic of Ireland:

INPUT, Dept AN, Marshall Cavendish Services,
Newtown Road, Hove BN3 7DN

Australia, New Zealand and Malta:

Back numbers are available through your local newsagent

COPIES BY POST

Our Subscription Department can supply your copies direct to you regularly at £1.00
each. For example the cost of 26 issues is £26.00; for any other quantity simply
multiply the number of issues required by £1.00. These rates apply anywhere in the
world. Send your order, with payment to:

Subscription Department, Marshall Cavendish Services Ltd,
Newtown Road, Hove, Sussex BN3 7DN

Please state the title of the publication and the part from which you wish to start.

HOW TO PAY: Readers in UK and Republic of Ireland: All cheques or postal orders
for binders, back numbers and copies by post should be made payable to:

Marshall Cavendish Partworks Ltd.


QUERIES: When writing in, please give the make and model of your computer, as
well as the Part No., page and line where the program is rejected or where it does
not work. We can only answer specific queries—and please do not telephone. Send
your queries to INPUT Queries, Marshall Cavendish Partworks Ltd, 58 Old
Compton Street, London W1V 5PA.

INPUT IS SPECIALLY DESIGNED FOR:

The SINCLAIR ZX SPECTRUM (16K, 48K, 128 and +),
COMMODORE 64 and 128, ACORN ELECTRON, BBC B
and B+, and the DRAGON 32 and 64.

In addition, many of the programs and explanations are also
suitable for the SINCLAIR ZX81, COMMODORE VIC 20, and
TANDY COLOUR COMPUTER in 32K with extended BASIC.
Programs and text which are specifically for particular machines
are indicated by the following symbols:

 **SPECTRUM 16K,
48K, 128, and +**  **COMMODORE 64 and 128**

 **ACORN ELECTRON,
BBC B and B+**  **DRAGON 32 and 64**

 **ZX81**  **VIC 20**  **TANDY TRS80
COLOUR COMPUTER**

SCORING AND TIMING ROUTINES

■	MINEFIELD GAME
■	KEEPING SCORE
■	ADDING A GAME TIMER
■	QUICK DRAW REACTION TIMER

There's nothing like knowing that you only have two seconds to get back to base or that you're within ten points of the high score to get your pulse racing. All arcade games keep some sort of count of the score or time to add to the excitement—and with a few simple programs, so can you.

Nearly all computer games need some kind of scoring and timing—or even both. Without them you can't judge how good you are at the game, or if you're improving, and there's little point in playing against your friends.

You could have someone sitting beside you counting your hits on the ferocious foe, but there's little point when you can program the computer to do it for you. A few more program lines will persuade the machine to remember high scores, too.

Similarly, there's no need to resort to a stopwatch for timing. As you saw in the maze game (pages 68 to 74), the machines all have a built-in timer, and it can be used in many ways to improve your games.

MINEFIELD

So that you can see how to set up scoring and timing routines in practice, here is a game, suitable for all the machines except the ZX81 and Vic 20, which will have the routines added to it progressively. Each routine is quite simple and could be added to other games, too.

The game is called Minefield and in it you are a tank commander whose mission is to rescue paratroopers who are recklessly landing in a minefield. Wherever the tank is moved it runs the risk of detonating a mine planted at random by the computer. As in a real minefield, the mines are invisible—so move with caution!

The tank (only a # mark, unfortunately, until you learn how to combine graphics and movement in a BASIC program) is controlled by using the usual movement keys—Z for left, X for right, P for up and L for down. Indeed, the core of the program is the 'moving around the screen routine' which you saw earlier (pages 54 to 59).



When you type in and RUN this section of the game you'll see that it's not yet complete—once you've rescued the paratrooper nothing happens except that the tank continues to wander around the screen aimlessly. The program has to be stopped by pressing the **BREAK**, **ESCAPE** or **RUN/STOP** keys, or you'll have to wait until the tank detonates a hidden mine. But don't worry, this will be improved as soon as you add the scoring and timing routines which follow.



On the Tandy, use 247 instead of 223 in Line 140 and 150; 251 instead of 239 in Line 160; 253 instead of 247 in Line 170.

```

50 LET PO = 210
60 CLS
70 PRINT@ 256, STRING$(32, "-")
90 LET X = RND(256) - 1
110 IF X <> PO THEN PRINT@ X, "O";
    ELSE GOTO 90
120 PRINT@ PO, "#";
130 LET LP = PO
140 IF PEEK(340) = 223 THEN
    LET PO = PO - 1:GOTO 190
150 IF PEEK(338) = 223 THEN
    LET PO = PO + 1:GOTO 210
160 IF PEEK(338) = 239 THEN
    LET PO = PO - 32:GOTO 220
170 IF PEEK(342) = 247 THEN
    LET PO = PO + 32:GOTO 220
180 GOTO 140
190 IF (LP AND 31) = 0 THEN
    LET PO = LP
200 GOTO 220
210 IF (PO AND 31) = 0 THEN
    LET PO = LP
220 IF PO > 255 OR PO < 0 THEN
    LET PO = LP:GOTO 140
230 PRINT@ LP, "□";
240 PRINT@ PO, "#";
250 LET M = RND(256) - 1
270 IF M = PO THEN PRINT@ PO, "□":
    PRINT@ 130, "BOOM!!—YOU'VE HIT
    A MINE":STOP
310 GOTO 130
  
```

The tank is moved around the screen by Lines 140 to 170, which check for the appropriate keypresses—see page 59. In order to confine the tank to the top half of the screen, Line 220 reads `IF PO > 255...` Screen location 255 is the last location on the top half of the screen, so the `IF... THEN` stops it straying below the dotted line drawn by Line 70.

Line 70 draws the line across the screen by using a new function—`STRING$. STRING$(32, "-")` as in Line 70 simply tells the Dragon to draw a string of 32 dashes. (If you

wanted ten question marks you'd use `STRING$(10, "?")`, and so on.)

The place that the paratroopers drop is chosen by the random number in Line 90, and Line 110 displays the parachute on the screen if the position chosen isn't already occupied by the tank. If it is, then Line 90 chooses another position.

Finally, the position of the hidden mine is chosen by Line 250. Line 270 then checks if the tank and mine are at the same screen location. If they are, the explosion message is displayed and the program stops.




```

15 VDU23;8202;0;0;0;
50 LET tankx = 20 : LET tanky = 10
60 CLS
70 PRINT TAB(0,13) STRING$(40, "-")
90 LET parax = RND(30) + 4
100 LET paray = RND(12)
110 IF (parax <> tankx) AND (paray <>
    tanky) THEN PRINT TAB(parax, paray)
    "O";ELSE GOTO 90
120 PRINT TAB(tankx, tanky) "#";
130 LET lasttankx = tankx : LET
    lasttanky = tanky
140 KEYS = GET$
145 IF KEYS = "P" THEN tanky = tanky - 1
150 IF KEYS = "L" THEN tanky = tanky + 1
160 IF KEYS = "Z" THEN tankx = tankx - 1
170 IF KEYS = "X" THEN tankx = tankx + 1
190 IF tankx < 5 OR tankx > 34 THEN
    LET tankx = lasttankx
200 IF tanky < 1 OR tanky > 12 THEN
    LET tanky = lasttanky
230 PRINT TAB(lasttankx, lasttanky) "□"
240 PRINT TAB(tankx, tanky) "#"
250 LET minex = RND(30) + 4
260 LET miney = RND(12)
270 IF (minex = tankx) AND (miney =
    tanky) THEN PRINT TAB(minex, miney)
    "□": PRINT TAB(7,7) "BOOM!!—
    YOU'VE HIT A MINE":END
310 GOTO 130
  
```

In the Acorn program, Lines 145 to 170 check which key has been pressed by the player before moving the tank. A full explanation of how this is done is on page 55. Lines 190 and 200 confine the tank to an area above the dotted line drawn by Line 70, and also stop the tank going off the top of the screen.

Line 70 draws the line across the screen using a new function—`STRING$. STRING$(40, "-")` simply tells the Acorn machines to draw a string of 40 dashes. (If you wanted a string of 15 plus signs, for example, you'd type `STRING$(15, "+")`.)

The place the paratroopers drop is chosen by Lines 90 and 100. The dropping position



is compared with the tank's position by Line 110. If the positions are different, then the paratrooper is displayed on the screen at the chosen position. If the tank's position is chosen then another dropping position is selected by going back to Line 90.

In a similar way Lines 250 and 260 choose a place for the mine. Line 270 then checks if

SCORING

In arcade-type games the score is normally increased when the screen positions of two objects are the same. The objects may be a missile and a target, a Pacman and a power pill, a tank and a paratrooper, a horse and a winning post or whatever your game demands.

So add these lines to the program to see how a scoring mechanism works in practice:



```
40 LET S = 0
280 IF X = PO THEN LET S = S + 1:
    GOTO 90
330 PRINT@295,S;"PARATROOPERS";
```



```
40 LET S = 0
280 IF (parax = tankx) AND (paray =
    tanky) THEN LET S = S + 1: GOTO 90
330 PRINT TAB(12,15);S;
    "□ PARATROOPERS"
```



```
40 LET s = 0
280 IF px = tx AND py = ty THEN LET
    s = s + 1: GOTO 90
330 PRINT AT 14,8;s;"□ PARATROOPERS"
```



```
40 LET S = 0
280 IF X = PO THEN LET S = S + 1:
    GOTO 90
330 POKE53280,254:POKE53281,246:
    PRINT "☠ □ □ □ □ □ □ □";S;
    "PARATROOPERS"
```

And change the STOP or END in Line 270 to GOTO 330. Line 270 should now read:



```
270 IF M = PO THEN PRINT@ PO,"□":
    PRINT@ 130,"BOOM!!—YOU'VE HIT
    A MINE":GOTO 330
```



```
270 IF (minex = tankx) AND (miney =
    tanky) THEN PRINT TAB(minex,miney)
    "!!": PRINT TAB(7,7)"BOOM!!—
    YOU'VE HIT A MINE":GOTO 330
```



```
270 IF mx = tx AND my = ty THEN
    PRINT AT my,mx;"!": PRINT AT
    8,3;"BOOM!!—YOU'VE HIT A
    MINE": GOTO 330
```



```
270 IF M = PO THEN POKE 1024 +
```

```
PO,32:PRINT Z$;"☠ BOOM!!—YOU'VE
    HIT A MINE":GOTO 330
```

Line 280 is the important one. It checks if the tank and the paratrooper are on the same screen location. If they are, then the score is increased by 1.

Line 40 sets the score to zero before the game starts and Line 330 displays the score. Changing the end of Line 270 makes the computer display the score after a mine has been detonated.

The player is now given a succession of paratroopers to rescue. When one is rescued another drops from the sky. The game stops when a mine is detonated by the tank moving across the same screen location.

HIGH SCORES

Adding a high score feature to a game isn't difficult. All you need is a variable for the high score—say, HS—and some way of updating it as the high score is exceeded, plus a display routine.

These are the lines you'll have to add to get a high score feature:



```
30 LET HS = 0
350 IF S > HS THEN LET HS = S
370 PRINT@ 424,"HIGH SCORE=";HS;
```



```
30 LET HS = 0
350 IF S > HS THEN LET HS = S
370 PRINT TAB(12,20) "HIGH
    SCORE=";HS
```



```
30 LET hs = 0
350 IF s > hs THEN LET hs = s
370 PRINT AT 18,8;"HIGH
    SCORE=";hs
```



```
30 LET HS = 0
350 IF S > HS THEN LET HS = S
370 PRINT "☠ □ □ □ □ □ □ □ □
    HIGH SCORE=";HS
```

First you must set the high score to the lowest level possible, so Line 30 sets HS at zero. After the game has stopped Line 350 compares the latest score (S) with the high score (HS). If the latest score is higher than the high score, then HS is updated so that it is the same as S. Finally, Line 370 displays the high score on the screen.

These lines will probably seem to be enough to add a high score to the game. Unfortunately this isn't true. Every time you



tell the computer to RUN the program it automatically forgets the value of HS, along with the values of all the other variables. To keep the value of HS you'll need to add the 'Do you want another go?' lines that have been used previously (page 35):



```
390 FOR F = 1 TO 1000:NEXT F
400 PRINT@ 130
410 PRINT@ 135,"ANOTHER GO?
    (Y/N)";
420 LET K$ = INKEY$:IF K$ = "" THEN
    GOTO 420
430 IF K$ = "Y" THEN GOTO 40
440 IF K$ = "N" THEN END
450 GOTO 420
```



```
390 FOR F = 1 TO 4000: NEXT F
400 PRINT TAB(7,7) STRING$(26,"□")
410 PRINT TAB(11,8)"ANOTHER GO? (Y/N)"
420 LET K$ = GET$
430 IF K$ = "Y" THEN GOTO 40
440 IF K$ = "N" THEN END
450 GOTO 420
```



```
390 PAUSE 100
400 PRINT AT 8,3; TAB31
410 PRINT AT 8,7;"ANOTHER GO? (Y/N)"
```




```
420 LET a$ = INKEY$
430 IF a$ = "y" THEN GOTO 40
440 IF a$ = "n" THEN STOP
450 GOTO 420
```



```
390 FOR F=1 TO 1000:NEXT F
410 PRINT "▣ □ □ □ □ □ □ □ □
ANOTHER GO? (Y/N)";
420 GET K$:IF K$ = "" THEN GOTO 420
430 IF K$ = "Y" THEN GOTO 40
440 IF K$ = "N" THEN PRINT "▣":
POKE 650,0:END
450 GOTO 420
```

This is what the extra lines do:

There is a short delay introduced by the FOR . . . NEXT loop in Line 390 on the Acorn, Commodore, Dragon and Tandy and by the PAUSE in the Spectrum program. Line 400 in the Acorn, Dragon, Tandy and Spectrum programs clears a portion of the screen ready for the message ANOTHER GO? (Y/N) to be PRINTed by Line 410.

The 'Another Go?' routine is on Lines 410 to 450—Line 430 restarts the program at Line 40 if Y is pressed, and Line 440 stops the program if N is pressed. Line 450 makes sure that any other keypress is ignored.

As there is no need now to type RUN every time you wish to have another play the value

of HS is preserved, although reLOADing the program will lose the value of HS even if you find some way of starting the program without typing RUN.

TIMING

As it stands the game depends too much on luck—the player just keeps on going until he runs over the concealed mine.

Some skill can be introduced into this kind of game by turning it into a race against the clock. By making these additions you can time how long it takes to rescue ten paratroopers.



```
80 TIMER = 0
290 IF S < 10 AND X = PO THEN GOTO 90
300 IF S = 10 THEN GOTO 320
320 LET T = TIMER
340 IF S = 10 THEN PRINT@ 327, "IN";
T/50; "SECONDS"
```



```
10 LET T = 0
80 TIME = 0
290 IF S < 10 AND (parax = tankx) AND
(paray = tanky) THEN GOTO 90
300 IF S = 10 THEN GOTO 320
320 LET T = TIME
340 IF S = 10 THEN PRINT TAB(12,16)
"IN□";T/100;"□SECONDS"
```



```
10 LET T = 0
80 POKE 23672,0: POKE 23673,0
290 IF S < 10 AND px = tx AND py = ty
THEN GOTO 90
300 IF S = 10 THEN GOTO 320
320 LET t = PEEK 23672 + 256 * PEEK
23673
340 IF S = 10 THEN PRINT AT 16,8;
"IN□";t/50;"□SECONDS"
```



```
80 LET TIS$ = "000000":L = 5
290 IF S < 10 AND X = PO THEN GOTO 90
300 IF S = 10 THEN GOTO 320
320 LET T = VAL(RIGHT$(TIS$,2)) +
60 * VAL(MID$(TIS$,3,2))
340 IF S = 10 THEN PRINT
"▣ □ □ □ □ □ □ □ □ IN";T;
"□SECONDS"
```

The timer in each of the machines is running all the time the computer is switched on. All you have to do to 'start' the timer, then, is to set the timer reading to zero. The line which resets the timer is Line 80—you'll notice that to reset the Acorn timer you enter TIME = 0, on the Commodore LET TIS\$ = "000000", on the Dragon/Tandy TIMER = 0 and on the Spectrum POKE 23672,0: POKE 23673,0. What the Acorn, Commodore, Dragon and Tandy lines are doing is clear, but Spectrum owners may be a little puzzled. The two POKEs set the values of two particular memory locations to zero. What happens is that the first memory location counts up to 255 before overflowing into the second.

The clock is 'stopped' by Line 320. The clock cannot actually be stopped, so you get the machine to remember what the reading was at a particular instant—when two objects coincide on the screen, say. In these programs the timer reading is called T—the Acorn says LET T = TIME, the Commodore LET T = VAL(RIGHT\$(TIS\$,2) + 60 * VAL(MID\$(TIS\$,3,2))), the Dragon/Tandy LET T = TIMER and the Spectrum LET t = PEEK 23672 + 256 * PEEK 23673. Again, the Spectrum method may be a little puzzling. It's looking directly at the values of the two memory locations which were set to zero earlier in the program. Location 23672, like any memory location in an 8-bit micro, can contain whole numbers between 0 and 255. Location 23673 goes up 1 every time location 23672 overflows.

So to work out the total amount of time that the Spectrum timer has been running you have to multiply the value in location 23673 by 256 and add on the value stored in location 23672.

You may now see why t is equal to PEEK



23672 + 256*PEEK 23673. (In any case PEEK and POKE will be dealt with fully in a later article.)

The Commodore clock works by updating a six character string as set up by Line 80. Starting from the right the string contains two digits for hours, then two for minutes and then two for seconds. Line 320 evaluates the string to get a reading in seconds. The number of minutes that have elapsed are multiplied by 60 and added to the number of seconds—MID\$(TI\$,3,2) looks at the number of minutes, and RIGHT\$(TI\$,2) looks at the number of seconds. (Again, a full explanation of string slicing comes in a later article.)

The clock must be stopped when the player has rescued ten paratroopers, so Line 300 checks if ten paratroopers have been rescued then jumps to Line 320 which 'stops' the timer. In the Commodore program Line 340

PRINTs the time when ten paratroopers have been rescued.

If a paratrooper has been successfully rescued *and* the total number rescued is less than ten, Line 290 makes another drop down.

Line 340 displays the rescue time *only* if ten have been recovered. The timer reading is divided by 50 in the Dragon, Tandy and Spectrum programs, and by 100 in the Acorn program. Thus the time displayed is in seconds—the Acorn clock is updated 100 times a second and the Dragon and Spectrum clocks are updated 50 times a second.

LOW TIME

Just as a high score was added to the game earlier, you might like to add a low time feature. In this particular game you'll be able to record the fastest time in which the ten paratroopers have been recovered, although

the principle can be applied to any timing you might wish to do.

These are the lines to add:



```
20 LET LT = 999999
360 IF T < LT AND S = 10 THEN LET
    LT = T
380 PRINT@ 452, "LOW TIME = "; LT/50;
    "SECONDS"
```



```
20 LET LT = 999999
360 IF T < LT AND S = 10 THEN LET LT = T
380 PRINT TAB(12,21) "LOW TIME = ";
    LT/100; "SECONDS"
```



```
20 LET lt = 999999
360 IF t < lt AND s = 10 THEN LET
```



```
lt=t
380 PRINT AT 21,4;"LOW TIME=";
lt/50;"□ SECONDS"
```



```
20 LET LT=999999
360 IF T < LT AND S=10 THEN LET
LT=T
380 PRINT"□□□□□□□□□□
LOW TIME=";LT;"□ SECONDS"
```

Just as when the high score was set up a very low 'high score' was originally set, now a ridiculously long 'low time' is set. Line 20 sets the low time variable (LT or lt) to a value of 999999.

Line 360 goes on to compare the latest time with the low time. If the latest time is shorter than the low time *and* ten paratroopers have been rescued, then the low time is altered so that it is the same as the latest time.

Finally, Line 380 displays the low time in seconds. The low time variable is divided by 50 or 100 to display the low time in seconds.

One thing you must remember is that if you are using a low time feature in a game you need to use the 'Another go?' routine otherwise you'll lose the value of the low time every time you RUN the program.

TIMING AND THE KEYBOARD

So far you've seen how to control the machine's timer from within a program by checking the positions of two objects on the screen. Another way to 'stop' the clock is to use the keyboard.

On page 54 you were shown how to use INKEY\$ or GET\$. They can be used just as easily to start or stop the timer as to control objects on the screen.

Here is a 'quick-on-the-draw' game which illustrates how the keyboard can be used to stop the timer:



```
20 CLS
30 LET N=RND(900)
40 FOR F=0 TO N
50 NEXT F
60 PRINT@ 269;"DRAW!!"
70 TIMER=0
80 IF INKEY$="" THEN GOTO 80
90 LET T=TIMER
100 PRINT@ 269;"BANG!!"
110 FOR F=1 TO 300
120 NEXT F
130 LET M=RND(25)
140 IF T < M THEN PRINT@ 264,
"YOU'VE SURVIVED"
150 IF T > M THEN PRINT@ 266,
"YOU'RE DEAD"
```

```
160 IF T=M THEN PRINT@ 264,
"YOU'RE BOTH DEAD"
```



```
20 CLS
30 LET N=RND(2000)
40 FOR delay=0 TO N
50 NEXT delay
60 PRINT TAB(17,13) "DRAW!!"
70 TIME=0
80 K$=GET$
90 LET T=TIME
100 PRINT TAB(17,13) "BANG!!"
110 FOR delay=1 TO 1000
120 NEXT delay
130 LET M=RND(40)
140 IF T < M THEN PRINT TAB(12,13)
"YOU'VE SURVIVED"
150 IF T > M THEN PRINT TAB(14,13)
"YOU'RE DEAD"
160 IF T=M THEN PRINT TAB(12,13)
"YOU'RE BOTH DEAD"
```



```
20 CLS
30 LET n=INT(RND*400)+1
40 PAUSE n
60 PRINT AT 11,14;"DRAW!!"
70 POKE 23672,0: POKE 23673,0
80 IF INKEY$="" THEN GOTO 80
90 LET T=PEEK 23672+256*PEEK
23673
100 PRINT AT 11,14;"BANG!!"
110 PAUSE 50
130 LET m=INT(RND*35)+1
140 IF t < m THEN PRINT AT 11,9;
"YOU'VE SURVIVED"
150 IF t > m THEN PRINT AT 11,11;
"YOU'RE DEAD"
```

```
160 IF t=m THEN PRINT AT 11,9;
"YOU'RE BOTH DEAD"
```



```
20 PRINT "□"
30 LET N=INT(RND(1)*900)+1
40 FOR F=0 TO N
50 NEXT F
60 PRINT "□□□□□□□□";TAB(16);
"DRAW!!"
70 LET TIS="000000":POKE 198,0
80 GET K$:IF K$="" THEN GOTO 80
90 LET T=TI
100 PRINT "□□□□□□□□";TAB(16);
"BANG!!"
110 FOR F=1 TO 300
120 NEXT F
130 LET M=INT(RND(1)*35)+1
140 IF T < M THEN PRINT
"□□□□□□□□" TAB(51)
"YOU'VE SURVIVED"
150 IF T > M THEN PRINT
"□□□□□□□□" TAB(53)
"YOU'RE DEAD"
160 IF T=M THEN PRINT
"□□□□□□□□" TAB(50)
"YOU'RE BOTH DEAD"
```

The program displays 'DRAW!!'—the player then has to press any key as quickly as possible. The reaction is timed from the moment of display and the player is told whether the computer or he has won.

The program is very simple. After Line 20 has cleared the screen a random pause is introduced by Lines 30 to 50. Line 60 displays 'DRAW!!' and the timer is started immediately by Line 70. Line 80 makes the machine wait until a key is pressed before continuing. The line is exactly as was used in 'Keyboard Control' (see page 54).

Immediately any key has been pressed, the program continues to Line 90 which effectively stops the timer by calling the timer reading T. Line 100 displays 'BANG!!'.

There is a pause introduced by Lines 110 and 120 (Line 110 only on the Spectrum) before a draw time is chosen for the machine—Line 130 does this.

Now the machine has two variables, your time T, and the machine's 'time' M. Lines 140 to 160 compare them and display whatever the outcome of the gunfight was.

If you are using an Acorn machine, don't be tempted to use INKEY to cause the delays at Lines 30 and 110 as you did in the earlier programs on animation. The problem now is that the delay is cut short if you press a key. So if you happen to keep your finger on a key, the computer will flash through this game in a fraction of a second.



Is there any limit to the period I can time?

Yes, there is—though it is usually so long that in practice it makes no difference. The timer in most home computers runs at a similar speed, and the limiting factor is how many time pulses the computer is able to remember. The Dragon, Tandy and Commodore can hold up to two bytes (65,535) which works out to about 22 minutes. The Spectrum runs to three bytes—nearly four days. The BBC and the Electron store the time in four bytes which means they can be set to count for almost 350 years!

ALL ABOUT READ AND DATA

Getting the computer to READ a store of DATA saves you typing in long programs which are slow to RUN. You can use the technique for anything from graphics to simple indexes.

One of the features that makes a computer so versatile is the use of variables. And for the most part, assigning values to them is fairly straightforward—you can simply say LET X=5, for example. But sometimes the amount of such information that you want to use can get out of hand. That is where DATA statements, and the accompanying READ and RESTORE commands, come in handy. Note that this is not available in ZX81 BASIC.

The word DATA has a very specific meaning here. 'Data' tends to be used to mean a variety of things. For example, a program, a machine code routine and an array stored on cassette tape or floppy disk are all called 'data'. But in this article it just means the DATA statement, and the items stored in it.

The first method of putting something into a variable that people come across—whether it is giving a value to a numeric variable, or entering a string—is through a simple INPUT statement. But that is only of use when information is being supplied to the computer by the user, each time the program is RUN.

Often, however, there are fixed values which do not need to be entered or altered by the operator, and so can be permanently incorporated into the program. This is, of course, where the LET statement comes in. But if there is a lot of information, using masses of LETs is tedious to program and slow to RUN.

Here's an example where a fixed list of headings might be used in a program for printing out a document:

```
10 LET A$ = "DAY"
20 LET B$ = "WEEK"
30 LET C$ = "MONTH"
40 LET D$ = "YEAR"
50 PRINT A$,B$,C$,D$
```

However, you can also use DATA statements to set them up instead.

```
10 READ A$,B$,C$,D$
20 PRINT A$,B$,C$,D$
100 DATA DAY,WEEK,MONTH,YEAR
```

If you now wanted to use not four but fifty different headings, the first program would require an extra 46 statements, the second only one or two, depending on the computer and the size of the headings.



Note that on the Spectrum the DATA words must be surrounded by quotation marks. So Line 100 is written as:—

```
100 DATA "DAY","WEEK","MONTH",
"YEAR"
```

HOW READ AND DATA WORK

How, in fact, do the READ and DATA statements work? When the computer comes across a READ instruction, it scans through the program until it comes across the first DATA statement. Then it assigns the first item of DATA to the variable in the READ command.

So in the program above, the DATA string "DAY" is assigned to the variable A\$, then "WEEK" is assigned to B\$ and so on.

DATA statements usually appear at the end of a program, where they do not 'get in the way' while the rest of the program is being

written and de-bugged. But in fact they can go anywhere at all. The computer simply ignores any DATA statement unless a READ statement makes it do otherwise. This next program would work perfectly well:

```
10 DATA GERMANY
20 READ A$,B$
30 DATA FRANCE,ITALY,SPAIN
40 READ C$,D$
```

Obviously, though, the program is easier for you to read if all the DATA is gathered together in one place. And there is one firm rule: the DATA *must* appear in the order that the computer will READ it.

The number of items a single DATA list can hold depends on the maximum line length of your computer. As soon as one line is full just start another—the computer still takes them in order.

- HOW READ AND DATA WORK
- DIFFERENT TYPES OF DATA
- AVOIDING PROBLEMS
- USING DATA TO MAKE A SIMPLE DIRECTORY

- THE RESTORE STATEMENT
- USES FOR DATA LISTS
- SELECTING FROM DATA
- HOW TO DRAW SIMPLE GRAPHICS FROM DATA



PROBLEMS WITH DATA

It is easy to make mistakes when entering DATA. The main problems occur when you do not have enough items or you try to READ in the wrong kind. If the DATA line above had been wrongly typed in as DATA 256,PARIS,a*5 then A\$ would be '256'. Although wrong, except on the Spectrum it would not be rejected.

It would be a different matter when the computer tried to READ the string 'PARIS' into the variable N. One of two things would normally happen. The computer might notice that the DATA was the wrong type and come back with an error message such as 'type mismatch' or 'bad data'. Alternatively, the computer might assume that PARIS was a variable name which it had not previously come across, and so would reply with something like 'variable not found'.

The other common error is supplying too few DATA entries. This often happens when you are using a loop containing the READ command, as in the program below. A slight mistake in the loop parameter, or an item of DATA accidentally left out, and the computer will try to READ past the end of the DATA list. The result is that the program stops with an error message like 'out of data'.

SIMPLE DIRECTORY

Here is a program that uses a loop to READ the DATA. It is a very simple telephone directory program. You enter the name of the person and the computer PRINTs out the number.



```

5 CLS
10 PRINT TAB(10,2) "TELEPHONE
  DIRECTORY"
20 INPUT TAB(5,10) "PLEASE ENTER
  THE NAME";R$
30 FOR J=1 TO 5
40 READ N$,T$
50 IF N$=R$ THEN PRINT TAB(5,12)
  N$;"S NUMBER IS: ";T$;END
60 IF N$="END" THEN PRINT TAB(5,12)
  R$;" IS NOT IN THE LIST"
70 NEXT J
500 DATA JENNY,62-4325,PETER,499-
  6262,DEBBIE,567-02206,MARK,44-
  9976,END,END
    
```

DIFFERENT TYPES OF DATA

So far the DATA statements have been strings, but they could just as well be numbers. The Spectrum and Acorn computers allow you to use variables and functions as well, so you could have a line looking something like this:

```
DATA PARIS, 256,a*5
```

(Again, you would need quotation marks around the "PARIS" on the Spectrum.)

The variables in the READ command must be placed in the same order as the relevant DATA items. So the DATA line above would be read with the command:

```
READ A$,N,X
```

Note that the first variable is a *string* variable to match the first DATA item. But the next two are *numeric* variables because they are READing numbers.



```

5 CLS
10 PRINT@ 71,"TELEPHONE DIRECTORY"
12 PRINT
20 INPUT "PLEASE ENTER NAME";R$
30 FOR J=1 TO 5
40 READ N$,T$
50 IF N$=R$ THEN PRINT@ 320,N$;"S
  NUMBER IS: ";T$;END
60 IF N$="END" THEN PRINT@ 320,R$;
  " IS NOT IN THE LIST"
70 NEXT J
500 DATA ETHEL,38-42765,HUBERT,
  997-403,ARCHIBALD,77-40628,
  GERTRUDE,5666-99994,END,END
    
```



```

5 PRINT " "
10 PRINT TAB(10)"TELEPHONE
  DIRECTORY"
20 INPUT "PLEASE ENTER THE
  NAME";R$
25 PRINT
30 FOR J=1 TO 5
40 READ N$,T$
50 IF N$=R$ THEN PRINT N$;"S NUMBER
  IS: ";T$;END
60 IF N$="END" THEN PRINT R$;" IS
  NOT IN THE LIST"
70 NEXT J
500 DATA TOM,21-23233,JOHN,01-32358,
  JOE,22-45785,JANE,78-87779,END,
  END
    
```



```

5 CLS
10 PRINT AT 2,6;"TELEPHONE
  DIRECTORY"
20 INPUT "PLEASE ENTER THE NAME",
  R$
30 FOR J=1 TO 5
40 READ N$,T$
50 IF N$=R$ THEN PRINT AT 10,1;N$;
  "S NUMBER IS: ";T$; STOP
60 IF N$="END" THEN PRINT AT
  10,3;R$;" IS NOT IN THE LIST"
70 NEXT J
500 DATA "JENNY","43-15263",
  "DICK","43-14123","JULIE",
  "13-65312","BUSTER","13-71824",
  "END","END"
    
```


Type in the program, but use the names and telephone numbers of your own friends in the DATA line. You can put in as many DATA items as you like but be sure to adjust the loop counter in Line 30.

Note that the DATA list is terminated with END,END. This is so Line 60 can check if the end of the list has been reached. If the program has read this far it means it hasn't found the name and it PRINTs out a message to tell you so. END (or "END" on the Spectrum) has to be entered twice because Line 40 reads in *two* DATA items at a time and you would get an 'out of data' error otherwise.

Also note that the telephone number is read as a string. This is because it contains a punctuation mark—the dash—and so cannot possibly be a number. Strings in DATA can contain spaces if you like but not commas. Since DATA items are separated by commas, the computer would treat the comma as the end of that item. If you do need to enter a DATA string that contains a comma then enclose the whole string in quotation marks. This would be necessary in a DATA line like this:

```
10 READ N$,A$,B$,C$
20 DATA ALBERT SMITH,"23, HIGH
   STREET", BRIGHTON,BT6 4ST
```

The first line of the address has to be written in quotes because of the comma after the

Microtip

Keeping track of your DATA

Whatever they refer to, DATA lines often have one thing in common—it takes a good deal of time to work them out and type them in. This is all very well the first time round, but you may well need to come back to a program much later.

Arranging the DATA neatly only takes a little longer, and can save you ages in the end. The best plan to follow depends on the DATA itself.

Where the DATA defines a block graphic or something similar, arrange the program lines to correspond directly to the graphic rows. If the DATA follows a repetitive format (such as entries in a telephone directory, for example), arrange each entry in exactly the same way in the program lines.



number 23. (You would have to do this on the Spectrum anyway, for all these strings.)

USING RESTORE

With the methods covered so far, a program can READ through a list of DATA only once, unless you reRUN the program. To reREAD DATA you must incorporate a RESTORE statement within your program.

For example, say you wanted to look up the numbers of several of your friends. The obvious thing to do is to put in an 'Another go' routine. First replace END or STOP in Line 50 with GOTO 80 then add the following program lines:



```
80 PRINT TAB(2,20)"DO YOU WANT
   ANOTHER NUMBER (Y/N)?"
90 K$ = GET$
100 IF K$ = "Y" THEN GOTO 5
110 END
```



```
80 PRINT AT 12,0;"DO YOU WANT
   ANOTHER NUMBER? Y/N"
90 PAUSE 0
100 IF INKEY$ = "y" THEN GOTO 5
110 IF INKEY$ = "n" THEN GOTO 2000
```



```
80 PRINT @32;"DO YOU WANT ANOTHER
   NUMBER(Y/N)?"
```

```
90 K$ = INKEY$:IF K$ = ""
   THEN GOTO 90
```

```
100 IF K$ = "Y" THEN GOTO 5
110 END
```



```
80 PRINT TAB(4)"DO YOU WANT
   ANOTHER NUMBER(Y/N)?"
90 GET K$:IF K$ = "" THEN GOTO 90
100 IF K$ = "Y" THEN GOTO 5
110 END
```

But what happens when you try to RUN the program? The first time through, the program works fine. But when you press a key for another go it fails due to lack of further DATA. This is because it came to the end of the DATA list on the first RUN. Fortunately there is an easy solution to this problem. Add:

```
15 RESTORE
```

This time the program keeps working every time. This is because the RESTORE command instructs the computer to go right back to the beginning of the DATA list.

When you are developing a program it is a good idea to put a RESTORE statement near the beginning. This stops you from running out of DATA on trial RUNs. If you are likely to want to take out the RESTORE line later, put a REMark statement after it to remind you.

By using RESTORE you can reuse a DATA list as often as necessary. This is particularly

USING RESTORE POINTERS

The only problem with DATA so far is that the information must always be called in the same sequence, starting from the same place. Using RESTORE, you can always go back to the beginning of a list even if you have not reached the end. But how do you jump into the middle of a list?

On Acorn computers, and the Spectrum, there is a way round this. Rather than using just one list, you can actually use several, directing the computer to the appropriate one. In the graphics program (right and overleaf) try adding these extra lines:



```
15 INPUT " <T> HIN OR <F> AT
    HOUSE ",H$
16 IF H$="T" THEN RESTORE 2000
17 IF H$="F" THEN RESTORE 1000
18 IF H$ <> "T" AND H$ <> "F" THEN
    GOTO 15
2000 DATA 400,100,900,100,900,600,
    400,600,400,100
2010 DATA 500,200,550,200,550,400,
    500,400,500,200,750,200,800,200
    800,400,750,400,750,200
2020 DATA 600,100,700,100,700,400
    600,400,600,100
2030 DATA 300,600,1000,600,650,800,
    300,600
```



```
6 INPUT "OLD OR MODERN BRIDGE?";
    y$
7 IF y$="o" THEN RESTORE 1000
8 IF y$="m" THEN RESTORE 2000
9 IF y$ <> "o" AND y$ <> "m"
    THEN GOTO 5
2000 DATA 83, 84, 85, 86, 87, 88, 89,
    90, 90, 90, 90, 90, 90, 90, 89,
    88, 87, 86, 85, 84, 83
2010 DATA 42, 55, 63, 65, 63, 55, 42
```

The numbers following the RESTORE commands are known as the RESTORE *pointers*. They direct the computer to a particular DATA list beginning at that line number. So in the Acorn program, if you press F for a fat house, the RESTORE pointer becomes 1000, in Line 17. In the Spectrum program, Line 7 has a similar effect.

In fact if you want the computer to go to the very first DATA list a pointer is not necessary. RESTOREing without a line number is the same as RESTOREing with the number of the first DATA line. Line 17 in the Acorn program could have been written as IF H\$="F" THEN RESTORE, with precisely the same effect.

Restore pointers are very useful in games programming. In a lander-type game, for example, you can write the program to test whether you have crashed or landed safely. The DATA lists can then contain information for a crash sound or a victory fanfare, and the RESTORE pointer can point to the correct list.

USING DATA FOR GRAPHICS

DATA statements are very useful in graphics programs, to fix the coordinates for plotting and drawing or to call up ROM graphics. Using DATA in this way is ideal for irregular shapes such as those overleaf, as many more program lines would be needed using individual printing or drawing instructions.

In some cases, though, DATA is not much use. And that is when you are plotting very regular shapes where the coordinates or the graphics shape can be calculated. There is an example of this in the Dragon and Tandy program where the castellations on the fort are calculated rather than listed out.

The DATA in the programs below has been split into several lines deliberately. This is because modifying DATA can be difficult unless clear-cut divisions are made—just imagine returning to a program at a later date and trying to match DATA entries to each of the variables. You can, of course, follow through each of the READ instructions, but this is tedious for a long program. So section off each group of DATA statements and—if you can spare the memory—use plenty of REMs to make things a little clearer.



The following program draws a house:

```
10 MODE 0
15 RESTORE
20 FOR N=1 TO 4
30 READ X,Y
40 MOVE X,Y
50 FOR M=1 TO 4
60 READ X,Y
70 DRAW X,Y
80 NEXT M
90 NEXT N
100 READ X,Y
110 MOVE X,Y
120 FOR P=1 TO 3
130 READ X,Y
140 DRAW X,Y
150 NEXT P
160 PRINT "PRESS ANY KEY FOR
    ANOTHER GO"
170 Q=GET
180 CLS:GOTO 10
1000 DATA 300,100,1000,100,1000,
    400,300,400,300,100
1010 DATA 400,200,500,200,500,300,
```



useful in programs where things like labels, or headings for tables, need to be repeated at different times. For example, you can put the months of the year in a DATA list for some sort of calendar program which makes repeated use of these in tables or graphs.

The RESTORE instruction is particularly useful when you want to keep READING through a list to search for a particular item—as the telephone directory program showed.

USES FOR DATA

DATA lists are extremely useful in all sorts of programs. You may have already used them for the user defined graphics (pages 8 to 15) and for drawing a maze (pages 68 to 74). You could also have the computer reading parameters for music or sound effects.

In adventure games, large numbers of DATA lines are frequently used to contain all the necessary text. Quizzes can store all the questions and answers in DATA lines. And arcade games in BASIC often use them to define both characters and backgrounds.

Experienced programmers use DATA for assembly language or machine code programs. Such a program may consist of a short loop containing POKE commands which READ from DATA lists.

In short, any program which contains standard texts, figures or functions can make good use of DATA lists. And with careful use, these lists can become a powerful programming tool.


```

400,300,400,200,800,200,900,200,
900,300,800,300,800,200
1020 DATA 600,100,700,100,700,300,
600,300,600,100
1030 DATA 200,400,1100,400,650,600,
200,400

```

The DATA gives the coordinates for the various parts of the house. Line 1000 gives the main outline, Line 1010 the windows, Line 1020 the door and Line 1030 the roof. The main part of the program just sets up the loops to READ in the relevant bits of DATA and draw the lines. Line 20 sets up the loop to draw four rectangles—the main house, the two windows and the door—and Line 50 sets the loop to draw four sides to each rectangle. Finally, Lines 120 to 150 draw the three sides to the roof.

Once the house is drawn you'll be asked if you want another go, and the program will repeat if you press a key—it doesn't matter which one. Notice the RESTORE in Line 15 to allow the program to rEREAD the DATA.

S

This program uses READ ... DATA to help draw a bridge. If you enter and RUN the program in stages it will be easier to see what's happening, and to check that your typing is correct. So:

```

10 FOR t=74 TO 80 STEP 3
20 PLOT 35,t
30 DRAW 175,0,-2.5
40 NEXT t

```

This bit uses a FOR ... NEXT loop to help PLOT three points near the left-hand side of the screen, then draws an arched line from each point. In Line 30, you'll recognise that the 175,0 means 'to a point 175 pixels to the right

of the starting pixel'. The -2.5 makes the line part of a circle, instead of straight.

```

100 FOR n=18 TO 33
110 READ a
120 PLOT n,a
130 DRAW 0,a
132 PLOT n+188,45
134 DRAW 0,a
140 NEXT n
1000 DATA 70,70,67,67,70,70,60,60,
57,57,60,60,57,57,60,60,70,70,67,
67,70,70

```

This is the section that draws the towers. The loop between Lines 100 and 140, plus the number 45 (pixels from the bottom) in Lines 120 and 132, PLOTs the bottom points of each of the closely-set vertical lines that draw the towers.

Then Lines 110 and 1000 take over. Line 110 tells the computer to READ, in Line 1000, the height (again in pixels) of each of the 22 vertical lines. So the first line is 0, 70—that is, vertical and 70 pixels high—the second line is 0, 70, the third 0, 67 and so on. If you want to watch it happening, try inserting a temporary 135 PAUSE 100 after Line 134.

```

300 PLOT 0,75
310 DRAW 255,0,-0.1
320 PLOT 0,78
330 DRAW 255,0,-0.1

```

These lines draw the roadway, the -0.1 producing a slight upward curve. And finally

```

400 FOR r=62 TO 182 STEP 20
410 PLOT r,78
420 READ b
430 DRAW 0,b
440 NEXT r
1010 DATA 42,55,63,65,63,55,42

```

These lines produce the vertical ties between the arch and the roadway, the FOR ... NEXT loop helping to PLOT the starting positions of the ties while Lines 420 and 1010 govern their height.

If you want the program to reRUN automatically, add these lines:

```

5 CLS:RESTORE
450 GOTO 5

```

Line 5 clears the screen and then allows the program to reREAD the DATA.





The program here uses a READ ... DATA routine to draw a country cottage from some of the ROM graphics on the Commodore 64:

```

10 PRINT"☐"
20 POKE 53280,3:POKE 53281,5
30 FOR A=1 TO 7
40 FOR B=1 TO 14
50 READ X
60 POKE 1024+291+(A*40)+B,X
65 POKE 55296+291+(A*40)+B,7
70 NEXT B
75 NEXT A
500 DATA 32,32,32,32,32,32,32,32,
32,32,108,160,123,32
505 DATA 32,32,233,160,223,32,32,
32,32,118,160,160,160,116
510 DATA 32,233,160,160,160,223,
32,92,32,160,160,160,160,160
520 DATA 32,102,102,102,102,102,
160,160,223,108,160,102,160,120
530 DATA 32,102,91,102,172,102,
102,91,102,32,32,102,32,32
540 DATA 32,102,102,102,160,102,
102,102,102,32,32,102,32,32
550 DATA 99,99,99,99,99,99,99,99,
99,99,99,99,99,99

```



1. A house on the Acorn from just four DATA lines printer 2. A Spectrum bridge using PLOT and DRAW, plus DATA

Seven lines of DATA are used: these correspond to the seven screen rows (lines) which form part of the FOR . . . NEXT loop started in Line 30. The fourteen entries within each DATA line correspond to the fourteen columns specified in the FOR . . . NEXT loop beginning in Line 40.

When the value of X—a screen display code (see your manual)—has been read as instructed in Line 50, it is POKEd into screen memory as directed by Line 60. A corresponding POKE has to be placed into colour memory for the character to show: this is achieved by Line 65. In both these lines, the number 291 starts the graphic at a point towards the centre of the screen, but the actual position is dictated by the values of the variables A and B. Now key in the following lines:

```
220 FOR K=1 TO 2000
230 NEXT K
240 PRINT "⏏️ ⏏️ PRESS ANY KEY
FOR ANOTHER GO"
250 GET K$:IF K$="" THEN GOTO 250
270 GOTO 10
```


RUN the program and respond to the prompt after the delay. Immediately you will see an 'out of data' error message: there is no more DATA for the reRUN Line 50 to access. However, by simply adding:

```
260 RESTORE
```

which sends the computer back to the start of the DATA statement, the program can be RUN any number of times.



This program will build up, in stages, a picture of a castle. Type in these lines and RUN them:



3. ROM graphics on the Commodore called up to order

```
10 PCLEAR4
20 PMODE4,1
30 PCLS5
40 SCREEN1,1
50 READ SX,Y
60 LINE-(SX,SY),PSET
70 FOR K=1 TO 18
80 READ X,Y
90 LINE-(X,Y),PRESET
100 NEXT K
270 GOTO 270
500 DATA 64,160
510 DATA 64,60,32,60,48,40,64,60,32,
60,32,160,110,160,110,120
520 DATA 152,120,152,160,228,160,
228,60,212,40,196,60,228,60
530 DATA 196,60,196,160,196,74
```


Lines 10 to 40 set up the high resolution graphics screen ready for drawing. Line 270 keeps it switched on.

Lines 50 and 60 are unusual. The coordinates of the start point of the castle are read, and a buff line is drawn on the buff background to the start point. Sometimes it's useful to be able to draw these 'invisible' lines, because they can join together visible lines in a continuous programming pattern.

The lines from 70 to 100 draw the outline of the castle. By the end of this piece of program 38 items of DATA have been read.

Now add these lines and you'll see some castellations when you RUN the program:

```
110 FOR K=1 TO 33
120 LET X=X-4
130 LINE-(X,Y),PRESET
140 IF Y=74 THEN LET Y=78 ELSE
LET Y=74
150 LINE-(X,Y),PRESET
160 NEXT K
```



4. The Dragon/Tandy castle is built up in stages

You're probably wondering what has happened to the READ line. This is one occasion when using READ and DATA will not save you work—you'd need 66 items of DATA to define the corners of the castellations, so this way—checking positions—saves you typing all that extra DATA.

These lines will draw some windows:

```
170 FOR K=1 TO 8
180 READ X,Y
190 LINE(X,Y)-(X+4,Y),PRESET
200 LINE(X+2,Y-2)-(X+2,Y+6),
PRESET
210 NEXT K
540 DATA 46,80,46,120,210,80,210,
120,86,90,170,90,80,132,178,132
```

Notice that you don't have to have DATA for each end of the windows, four sets in all. As they are all the same size only one set of DATA is needed per window.

Now try adding these lines. RUN the program and see what happens.

```
220 FOR K=1 TO 4000
230 NEXT K
240 CLS:PRINT@33,"PRESS ANY KEY
FOR ANOTHER GO"
250 LET IN$=INKEY$:IF IN$=""
THEN GOTO 250
270 GOTO 30
```

When you press a key you will get OD or 'out of data' error. The reason for this is that the program has finished the list of DATA—there is no more. Fortunately you don't have to type in all the DATA again. Just add this line which makes use of the RESTORE statement to instruct the computer to go back to the start of the DATA list.

```
260 RESTORE
```


LEARNING TO COUNT ON ONE FINGER

Computers count in twos—it's as if they had millions of hands but only one finger on each. And for effective machine code programming, it's a system you need to learn as well

One of the problems of learning machine code is that you need to understand a little about the theory of numbers. This is not quite as daunting a task as it may seem. If you can count up to 16 you should have no difficulty. But first of all you have to learn how to count up to two.

WHY BASE TEN?

Even the most unmathematical person has no problem telling the time, working out their change or keeping track of football scores. The use of numbers is such a part of everyday life that we never give a thought to how they work. But when you start programming in machine code you need to take a closer look at the way numbers work.

In the western world we use a number system based on the number 10. That means we start counting in the digits 0 to 9. When we need a number higher than 9, we have used all the available digits. So we put a 1 in the next place to the left and a 0 in the right-hand place. Ten is not represented by a single digit, but by two digits we've used before.


If we go on counting, we go through the digits 0 to 9 again in the right-hand place. And when we've reached 9 and add one more, 1 is again clocked up in the next place to the left. And when that place has filled up to 9 and one more has to be added, a 1 appears in the next place to the left and the two right-hand places reset to 0.

This is called a *base-ten*, or *decimal*, number system, because the value of the digit is increased by a factor of ten in each place as you count from the right. For example, the number 3,275 is worth $5 + 7 \times 10 + 2 \times 10 \times 10 + 3 \times 10 \times 10 \times 10$ —or $5 + 70 + 200 + 3,000$. Each digit increases its worth by ten times its face value for every place it is moved to the left.

All this seems very commonplace because we do it every day without thinking about it. So much so that, in base ten, it seems self-evident. But we don't always use base ten.

ANCIENT AND MODERN

The ancient Babylonians had a system of numbers based on a unit of 60. The vestiges of this can still be seen in our measures of time



and angles. There are 60 seconds in a minute, 60 minutes in an hour, 60 minutes of arc in a degree and six times 60 degrees—in other words 360—in a complete circle.

You can count up to 59 seconds, but if you add one more you clock up 1 minute and start counting the seconds from 0 again. And when you have 59 minutes and 59 seconds, one more second will give you an hour, while the minutes and seconds places go back to 0 before starting to count up again.


The remnants of number systems with other bases can be seen, too, in the old

imperial system of measurement. There are eight pints in a gallon, eight furlongs in a mile, eight stones in a hundredweight and 16 ounces in a pound. Twelve is another base that occurs often. There are 12 inches in a foot, 12 old pennies in a shilling, 12 whatever's in a dozen and 12 dozens in a gross.

Base 12 used to be used in British coinage, and for goods in dozens and grosses, because it was easy to divide by 2, 3, 4 and 6. Money and goods often have to be shared among more than one person and the base 12 made these transactions much easier to perform.

■	COUNTING IN BASE TEN
■	OTHER NUMBER SYSTEMS
■	HOW TO COUNT IN BASE NINE
■	DOING SUMS IN DIFFERENT BASES

■	A CALCULATOR PROGRAM FOR ALL NUMBER BASES
■	THE BINARY SYSTEM
■	BITS AND BYTES
■	HOW THE COMPUTER ADDS UP



1. Our ten fingers lead us to count in tens, and counting aids like the simple abacus or bead counter are a mechanical model for our fingers.

would be able to count up through the digits from 0 to 8 in the normal way. But if we added one more to 8 it would clock up the digit 1 in the next place to the left and return the first place to 0. In other words the decimal number nine would be represented in nonary by 10.

Again if 1 was added to 18 you'd get 20, and if 1 was added to 88 you'd get 100.

As you can see from the example below, the ordinary rules of arithmetic apply whatever number you base your number system on. You can try simple sums in an eight-based or seven-based system, for example. Whatever base you choose to do calculations in, the ordinary rules of arithmetic still apply.

In our nonary system, for instance, as you move each place to the left the face value of the digit is increased by a factor of nine. So 3,275 in nonary is equal to $5 + 7 \times 9 + 2 \times 9 \times 9 + 3 \times 9 \times 9 \times 9$ or 2,417 in decimal. But, you'll note, there is no digit 9 in the nonary system. Nine is represented by 10. So it would be just as correct to say that 3,275 is equal to $5 + 7 \times 10 + 2 \times 10 \times 10 + 3 \times 10 \times 10 \times 10$ in nonary.

With a little bit of mental agility you should be able to do simple sums in nonary. For example, 99×9 equals 891 in decimal. In nonary, 99 is represented by 120—that is $1 \times 9 \times 9 + 2 \times 9$ or $81 + 18$ —and 9 is 10.

So in nonary 99×9 translates into 120×10 which is 1,200 or $1 \times 9 \times 9 \times 9 + 2 \times 9 \times 9$ or 891 in decimal.

Any addition, subtraction, multiplication or division of nonary numbers will work. Check for yourself. But you have to be a bit careful. Remember, for example that 16 in decimal is 17 in nonary, if you carry or borrow one in nonary it is only worth nine, and in division, 3 goes into 10 three times.

PROBLEMS OF THE OVER-TENS

Dealing with number systems with bases greater than ten is a little more difficult though. To handle them easily you have to devise new digits. For example, if you were to

use a twelve-based number system you would need to invent digits to represent ten and eleven. In duodecimal, 10 would mean 1×12 , or 12. And 11 would mean $1 \times 12 + 1$ —that is, 13 in decimal.

The easiest way to extend the range of digits is to use the alphabet. Ten could be A, say, and eleven could be B. Some of your duodecimal numbers would look like this: A2, BA and 7B—or 122, 142 and 95 in decimal.

Microtip

A model for numbers in different bases

It can be hard to visualize just how a different number system works, because everyone is so used to counting in tens. After all, it isn't easy to chop off some fingers, or grow extra ones, so you can count on them! But you can, for example, tape two fingers together to try base nine, or carry two extra tally sticks to work in base twelve, and so on.

But a better model is to make your own bead counter—similar to the ones shown on these pages. Just as an ordinary bead counter or abacus can help to clarify and speed up conventional arithmetic in base 10, so you can make a bead frame with as many counters on each line as you want. You don't even need to make a proper frame—just an arrangement of grooves in which you can roll the beads back and forth would be adequate.

Don't forget that you need one less bead in each row than your number base. So in base ten you have nine beads and binary only needs one. No beads slid across equals zero, one bead slid across equals one and so on. When you get beyond nine, you 'carry' a one to the next row, and slide back all the beads to zero.

Ten, on the other hand, is divisible only by two numbers—2 and 5.

It is often convenient to use a number system with a base other than ten. For example, a pound in weight can be broken down into ounces by repeated halving.

NINE OF THE BEST

There is no reason, in fact, why you can't use a system based on any number you like. For example, if we had all been born with one finger fewer on one hand, we should probably have a nonary—or nine-based—system. We

AN ALL-BASE CALCULATOR

Doing arithmetic in bases other than ten can lead to severe brain strain. So here is a program that will do arithmetic in any base up to 36. When you RUN it, it will ask you which base you want to use. Key the appropriate number in, in decimal.

Next it asks you for a number. This should be entered in the base you have selected. If you've chosen base eight, you shouldn't use any digit over 7, for example. And if you've picked a base greater than ten, digits over 9 should be keyed in using the letters of the alphabet, in capitals. Ten will be A, eleven B, twelve C and so on.

Then the computer will ask you for an arithmetic sign. Key in +, -, * or /. After that you have to supply another number in the same base as before.

If your sum is a division and the answer is not an integer, the computer will tell you that it will not work. 'Won't go' will be PRINTed on the screen. Otherwise the sum and its answer, in the base you've selected, will be displayed on the screen.



```

20 INPUT "BASE (UP TO 36)";B%
30 INPUT "NUMBER (INTEGER)";A$
40 INPUT "SIGN";S$
50 INPUT "NUMBER (INTEGER)";B$
60 PRINT " "
70 P$ = A$:GOSUB 200
80 DA = D%
90 P$ = B$:GOSUB 200:DB = D%
100 IF S$ = "*" THEN X% = DA*DB
102 IF S$ = "+" THEN X% = DA + DB
104 IF S$ = "-" THEN X% = DA - DB
106 IF S$ = "/" THEN X% = DA/DB
120 ANS = ""
130 IF B%*(X%/B% - INT(X%/B%)) > 9
  THEN ANS = CHR$(55 + B%*(X%/B% -
  INT(X%/B%))) + ANS:GOTO 150
140 ANS = RIGHT$(STR$(B%*(X%/B% -
  INT(X%/B%)),1) + ANS
150 X% = INT(X%/B%)
160 IF X% > 0 THEN GOTO 130
180 PRINT "BASE";B%
  " "A$S$B$ " = "ANS
190 END
200 D% = 0:IX% = 0
205 Y = ASC(RIGHT$(P$,1))
210 IF Y < 58 THEN D = Y - 48:
  GOTO 230
220 D = Y - 55
230 D% = D% + D*B%↑IX%:
  P$ = LEFT$(P$, LEN(P$) - 1):
  IX% = IX% + 1
240 IF LEN(P$) > 0 THEN GOTO 205
250 RETURN
  
```



On the ZX81, type in the following program entirely in capitals. Omit Line 10 completely, omit LINE in Lines 30, 40, and 50, and use ** instead of ↑ in Line 220.

```

10 POKE 23658,8
20 INPUT "BASE (UP TO 36)";b
30 INPUT "NUMBER (INTEGER)";LINE a$
40 INPUT "SIGN";LINE s$
50 INPUT "NUMBER (INTEGER)";LINE b$
60 CLS
70 LET p$ = a$:GOSUB 180
80 LET x$ = STR$ dec
90 LET p$ = b$:GOSUB 180
95 LET y$ = STR$ dec
100 LET z$ = x$ + s$ + y$: LET x = VAL z$
110 IF x < > INT x THEN LET n$ = "WILL
  NOT GO":GOTO 160
120 LET n$ = ""
130 LET u = INT (x/b): LET u = u*b: IF
  x - u > 9 THEN LET n$ = CHR$
  (55 + (x - u)) + n$:GOTO 140
135 LET n$ = STR$ (x - u) + n$
140 LET x = INT (x/b)
150 IF x > 0 THEN GOTO 130
160 PRINT AT 10,0;a$ + s$ + b$;" = ";n$
170 STOP
180 LET dec = 0
190 LET index = 0
200 LET y = CODE p$(LEN p$)
210 IF y < 58 THEN LET d = y - 48:
  GOTO 220
215 LET d = y - 55
220 LET dec = dec + d*b↑index
230 LET p$ = p$(1 TO (LEN p$) - 1)
240 LET index = index + 1
250 IF LEN p$ > 0 THEN GOTO 200
260 RETURN

10 MODE6:PRINT"
20 INPUT "BASE (UP TO 36)";B%
30 INPUT "NUMBER (INTEGER)";A$
40 INPUT "SIGN";S$
50 INPUT "NUMBER (INTEGER)"; B$
60 CLS:PRINT"
70 P$ = A$:GOSUB180
80 DECA$ = STR$(DEC%)
90 P$ = B$:GOSUB180: DECBS$ =
  STR$(DEC%)
100 X% = EVAL(DECA$ + S$ + DECBS$)
110 IF X% < > EVAL(DECA$ + S$ + DECBS$)
  THEN ANS$ = "WON'T GO":
  GOTO 160
120 ANS$ = ""
130 IFX%MODB% > 9THEN ANS$ = CHR$
  (55 + X%MODB%) + ANS ELSEANS$ =
  STR$(X%MODB%) + ANS
140 X% = X%DIVB%
  
```

```

150 IFX% > 0 GOTO 130
160 PRINTTAB(0,12) A$ + S$ + B$ +
  " = " + ANS$
170 END
180 DEC% = 0
190 INDEX% = 0
200 LET Y = ASC(RIGHT$(P$,1))
210 IF Y < 58 THEN D = Y - 48
  ELSE D = Y - 55
220 DEC% = DEC% + D*B% ^ INDEX%
230 P$ = LEFT$(P$,LEN(P$) - 1)
240 INDEX% = INDEX% + 1
250 IF LEN(P$) > 0 THEN GOTO 200
260 RETURN
  
```



```

10 CLS
20 INPUT "BASE (UP TO 36) ";B
30 INPUT "NUMBER (INTEGER) ";A$
  
```

2. Any number system can be modelled mechanically—the clock is in base 60, the bead frame in base 10




```

40 INPUT"SIGN";SS
50 INPUT"NUMBER (INTEGER)";B$
60 P$ = A$:GOSUB 210
70 X$ = STR$(DE)
80 P$ = B$:GOSUB 210
90 Y$ = STR$(DE)
100 IFSS = "*" THEN X = VAL(X$)
    *VAL(Y$)
110 IFSS = "/" THEN X = VAL(X$)
    /VAL(Y$)
120 IFSS = "+" THEN X = VAL(X$)
    + VAL(Y$)
130 IFSS = "-" THEN X = VAL(X$)
    - VAL(Y$)
140 IFX <> INT(X) THEN NS = "WON'T
    GO":GOTO190
150 U = B*INT(X/B):IF X - U > 9
    THEN NS = CHR$(55 + (X - U))
    + NS:GOTO170
160 NS = MID$(STR$(X - U) + NS,2)
170 X = INT(X/B)
180 IFX > 0 GOTO150
190 PRINT@257,A$ + "□" + SS +

```

```

"□" + B$;" = □";N$
200 END
210 DE = 0
220 IN = 0
230 Y = ASC(RIGHT$(P$,1))
240 IFY < 58 THEN D = Y - 48:GOTO260
250 D = Y - 55
260 DE = DE + D*B ↑ IN
270 P$ = LEFT$(P$,LEN(P$) - 1)
280 IN = IN + 1
290 IFLEN(P$) > 0 THEN GOTO230
300 RETURN

```

If you find the whole idea of doing arithmetic in different bases more than you feel like handling, note that even the computer cheats. This program actually converts the numbers you've INPUT into decimal, does the arithmetic, then converts the answer back into the base you've selected. Unlike the flexible human mind, your computer can do arithmetic only in decimal in a BASIC program. (This, despite converting the numbers into

binary when it translates the instructions into machine code to perform them!)

AND NOW, BINARY

In computers the most convenient number system to use is that based on two. This is because a computer is composed of electronic switching circuits which have two obvious states—off and on. The off state represents the digit 0 and the on state the digit 1. And in base two that is all the digits you need.

The base-two number system is known as *binary*. It is composed of nothing but 0s and 1s, all the other digits have been abolished. So if you start counting from 0 you can get as far as 1 in the normal way. Add 1 to 1, and 1 is clocked up in the next place to the left while the first place returns to 0. So, in binary, $1 + 1 = 10$.

Counting through from zero to eight gives 0, 1, 10, 11, 100, 101, 110, 111, 1000. Again these numbers obey all the ordinary laws of arithmetic. For example: if you add $10 + 11$, you first add the two right-hand digits, $0 + 1$. Then you add the two left-hand digits, $1 + 1$ which gives two, but two in binary is 10. So $10 + 11 = 101$ —in other words, $2 + 3 = 5$.

Subtraction in binary is a similarly straightforward process. The only thing to watch is when to carry.

The beauty of binary is that even for human beings multiplication and division are extraordinarily easy. At each stage of a multiplication you are either multiplying 0 by 0 (which gives 0), 0 by 1 (which again gives 0) or 1 by 1 (which gives 1).

In binary, in fact, multiplication is so easy that even a computer can do it.


Long division is similarly simple. At each stage you only have to decide whether the divisor goes into the number that is being divided once or no times.

Check for yourself that all the normal arithmetic processes work in binary. The panel on the next page gives a series of worked examples of addition, subtraction, multiplication and division for you to try.

BITS AND BYTES

Binary numbers accurately reflect what is going on in the computer. Every instruction or piece of DATA you feed into the computer is encoded in binary which the switching circuits of the computer manipulate and store. When you understand binary you begin to understand how the computer works.

Each of the digits of a binary number is represented electronically in the computer by a circuit that is on or off. If it is on, the value is 1. If it is off, the value is 0. Each digit circuit is known as a bit.



To memorize and manipulate binary numbers these circuits are organized into larger entities which can represent larger, more useful numbers. In nearly all home computers the bits are organized into groups of eight, giving what is known as an eight-bit byte. That means each byte can represent eight binary digits, so it can store any number between 00000000 or zero, and 11111111, or 128 + 64 + 32 + 16 + 8 + 4 + 2 + 1 which is 255. Numbers larger than 255 are represented by two bytes or more.

BINARY FRACTIONS

All the programs and examples shown so far have all dealt with whole numbers, but it is also possible to convert fractions into binary numbers.

For example, 1/2 is 0.1 in binary, 1/4 is 0.01, 1/8 is 0.001. You may already be able to see a pattern emerging. In fact, any fraction can be made up of 1/2s, 1/4s, 1/8s, 1/16s and so on—just as any whole number can be made up of 1s, 2s, 4s, 8s etc. It's just that it is more difficult to work out the binary equivalent of fractions because the series of 0s and 1s is often infinite!

When you're converting fractions, you can think of the 1/2s, 1/4s etc. as .5, .25 and so on. Then you can fairly easily work out that a fraction such as .75 is made up of 1*.5 + 1*.25, which means its binary equivalent is 0.11.

The next program gets the computer to work them out for you:

```

RT
10 CLS
    
```

```

20 PRINT@35, "DECIMAL TO BINARY
  CONVERSION"
30 PRINT: INPUT "ENTER A NUMBER
  BETWEEN 0 AND 1 ";N
40 IF N <= 0 OR N >= 1 THEN 30
50 NS="0."
60 FOR T=1 TO 30
70 N=N*2
80 NS=NS+CHR$(48+INT(N))
90 N=N-INT(N)
100 NEXT T
110 PRINT@257, "THE BINARY NUMBER
  IS:—"
120 PRINT: PRINTNS: PRINT
130 PRINT "ANOTHER NUMBER (Y/N) ?"
140 AS=INKEY$
150 IF AS="Y" THEN 10
160 IF AS <> "N" THEN 140
170 END
    
```



```

20 PRINT "DEC TO BIN
  CONVERSION"
30 PRINT: INPUT "ENTER A NUMBER
  BETWEEN 0 - 1";INPUT N
40 IF N <= 0 OR N >= 1 THEN 20
50 NS="0."
60 FOR T=1 TO 32
70 N=N*2
80 NS=NS+CHR$(48+INT(N))
90 N=N-INT(N)
100 NEXT T
110 PRINT "THE BINARY NUMBER
  IS:—"
120 PRINT " "NS
130 PRINT "ANOTHER NUMBER
    
```

```

(Y/N) ?"
140 GET AS
150 IF AS="Y" THEN RUN
160 IF AS <> "N" THEN 140
170 PRINT " " : END
    
```




```

10 CLS
20 PRINT "DECIMAL TO BINARY
  CONVERSION"
30 INPUT "ENTER A NUMBER
  BETWEEN 0 AND 1",N
40 IF N <= 0 OR N >= 1 THEN GOTO 30
45 PRINT "DECIMAL: - " :N
50 LET NS="0."
60 FOR T=1 TO 16
70 LET N=N*2+1E-9
80 LET NS=NS+CHR$(48+INT N)
90 LET N=N-INT N
100 NEXT T
110 PRINT "THE BINARY NUMBER
  IS:—"
120 PRINT NS
130 PRINT "ANOTHER NUMBER
  (Y/N) ?"
140 LET AS=INKEY$
150 IF AS="Y" THEN GOTO 10
160 IF AS <> "N" THEN GOTO 140
170 STOP
    
```




```

10 MODE 1
20 PRINT "DECIMAL TO BINARY
  CONVERSION"
30 INPUT "ENTER A NUMBER
  BETWEEN 0 AND 1",N
    
```



1. Binary addition works from the right. 0 + 0 gives 0. 0 and 1 give 1. 1 and 1 give 10, which is 0 and carry 1. Check how this works through the sum above

2. Binary subtraction is similarly simple. 1 - 1 gives 0. 0 from 1 gives 1, while if you try to take 1 from 0, you must carry this over to the next digit to the left




```

40 IF N <= 0 OR N >= 1 THEN 30
50 N$ = "0."
60 FOR T=1 TO 32
70 N = N*2
80 N$ = N$ + CHR$(48 + INT(N))
90 N = N - INT(N)
100 NEXT
110 PRINT "THE BINARY NUMBER IS :-"
120 PRINT N$
130 PRINT "ANOTHER NUMBER (Y/N)?"
140 A$ = GET$
150 IF A$ = "Y" THEN 10
160 IF A$ < "N" THEN 140
170 END
    
```


The program accepts any number between 0 and 1 and then PRINTs out its binary equivalent. Line 70 starts by doubling your number then Line 80 builds up the binary number one character at a time. The computer first works out the value of INT(N), which will be either 0 or 1, then this is added to 48 to make either 48 or 49—which are simply the ASCII codes for "0" and "1". All this is doing in effect is to turn the numbers into strings so they can be joined on to N\$. Line 90 takes the integer away from N to leave just the fractional part.

The Sinclair program is slightly different because while INT 1 is 1 as you would expect, INT (.5*2) comes out as 0. This is because it stores .5*2 as .9999999... So a small fraction has to be added on to N in Line 70 to tip the balance and make INT work as it should.

3. A binary abacus only needs single counters—in a computer, pulses of electricity in a circuit can do the job




3. At each stage of a binary multiplication, you either multiply 0 by 0, 0 by 1, which also gives 0, or 1 by 1, which gives 1. See how this applies in the example above



4. In binary long division you have to check whether the divisor goes into the number you are dividing once, in which case you enter 1, or no times, when you enter a 0

FROM THE ABACUS TO ELECTRONICS



1 0 1 1 1 0 1 1 - 0 1 0 0 1 1 0 1 = 0 1 1 0 1 1 1 0



0 1 1 1 1 1 1 + 0 0 0 0 0 0 0 1 = 1 0 0 0 0 0 0 0



1 1 0 0 1 1 1 1 0 1 1 1 1 0 - 0 1 1 1 0 1 1 0 0 0 1 1 0 0 1 1 = 0 1 0 1 1 0 0 1 1 0 1 0 1 0 1 1

Binary numbers can be added, subtracted, multiplied and divided like any other numbers, but were chosen for use in computers for their unique property—it is easy to tell if a circuit is on, or if it is off.

Binary digits are represented by computer circuits. When a circuit is off it represents a 0, when it is on it represents a 1. In the chips in your home computer the binary circuits are arranged in eights so eight-bit binary numbers can be represented. This gives a range of 00000000, or 0 in decimal, to 11111111, or 255 in decimal. If

you look through your computer's manual you will find the number 255—or 256 if you start counting from 1 instead of 0—comes up over and over again.

For some purposes larger numbers are required and two eight-bit binary memory locations are taken together to represent a 16-bit binary number. The Dragon microprocessor can handle these and sometimes does its arithmetic in 16-bit binary. In this case, numbers between 0000000000000000 and 1111111111111111, or 0 and 65,535 in decimal, can be handled.

SPOT-ON SCREEN DISPLAYS

Good screen display makes your instructions readable and your tables understandable. It's all a matter of using the right commands

PRINT and INPUT are two of the first commands that programmers learn. They are used all the time and in all sorts of programs and you should have a fairly good idea of how they work. But you may not be using them to their full advantage.

There are many ways of laying out text on the screen, and it is important to make a good display, especially if you write programs that you want other people to use. If the screen is too full of confusing instructions or the text is all jumbled up, then many people will become impatient and won't take the trouble to read it. Your carefully worked out quiz, chart or

game will then be useless.

So whatever sort of program you are writing, the text and instructions must be easy to read, and if you want someone to INPUT information it must be obvious exactly what you want.

Each computer has its own way of positioning text on the screen. The commands are usually TAB, AT, the @ symbol, or the cursor control characters. Punctuation—commas, semicolons and apostrophes—are also used to separate different PRINT statements.

This article explains how to make full use of all these commands so you can make neat and readable displays that are understandable at a glance.



You probably know that a command like this:

- USING TAB WITH PRINT AND INPUT
- POSITIONING WITH PUNCTUATION MARKS
- MAKING A NEAT DISPLAY
- HOW TO USE SPACING

PRINT "Good morning"

will PRINT the message on the first clear line starting at the far left. If you clear the screen first:

```
10 CLS
```


```
20 PRINT "Good morning"
```

it will PRINT it on the top line of the screen. The computer is doing just the same as you would if you were given a clean sheet of paper.

But you can tell the computer to PRINT the message at a different position. Add this line to the last program:

```
30 PRINT TAB(20) "Good afternoon"
```

This PRINTS out the words starting at column 20, halfway across the screen. Each word can be positioned separately if you want, using two TAB statements.





Try changing Line 30 to:

```
30 PRINT TAB(20) "Good" TAB(30)
   "afternoon"
```

This now PRINTS the first word at column 20 and the second word at column 30.

Note that there is *no* space between the keyword TAB and the first bracket. If you did leave a space the computer would not understand what you wanted and would PRINT a 'no such variable' error.

If you want the message PRINTed on a different line then you must put two numbers in the brackets. For example, this next line PRINTs the words at column 20 as before, but this time on line 15:

```
40 PRINT TAB(20,15) "Good night"
```

TAB with just one number is useful after you've already used one TAB command to move to a different line. For instance, to position two words (or numbers) on a single line, first move to the first item with TAB(column, line) then PRINT the second item at TAB(column)—you are already on the correct line.

But note: in this case the number in the second TAB is the number of columns from

the last one, *not* the absolute position on the screen. So this line from a program is quite acceptable:

```
50 PRINT TAB(20,17) "first" TAB(8)
   "second"
```

The word "second" is PRINTed 8 columns after the word "first", at column 28, not at column 8 as you might think at first.

If you tell the computer to PRINT beyond the end of the line—TAB(85) for instance—it just counts on to the next line or even several lines until it has moved the correct number of columns.

It is worth experimenting with the TAB command to get a better feel for what it will and will not do.

INPUTTING INFORMATION

TAB can be used with INPUT as well as with PRINT. Here's an example:

```
10 INPUT TAB(1,5) "Please enter your
   name□", name$ TAB(1,6) "and
   your age□", age
20 PRINT TAB(1,10) "Name□"; name$;
   "□ age□"; age
```

This will PRINT the first instruction at column

1, line 5 and then prompt you with a question mark for an INPUT. Whatever you type is stored as 'name\$'. The next instruction is PRINTed on the next line exactly below the first and another question mark prompts you for the second INPUT which is stored as 'age'. This is a numeric variable so you must type in a number. Line 20 simply PRINTs out name and age so you can see that the information has actually been stored.

If you don't want the question mark, simply miss out the commas in Line 10. This would be necessary in a line such as:

```
INPUT "Cost of item £" cost
```

as a question mark would spoil the display. 'Cost of item £67.45' is obviously much better than 'Cost of item £?67.45'.

Here's a short program that makes use of INPUT TAB. It allows you to enter cricket scores and automatically tabulates them into neat columns on the screen. There's no need to work out the batsman's average for the last column—the computer does this for you and PRINTs out the result as soon as you have entered the other figures.

```
10 CLS
20 PRINT TAB(1) "NAME" TAB(13)
```



```

"INNINGS" TAB(22) "RUNS" TAB(29)
"N.O." TAB(35) "AV"
30 LET ROW=3
40 INPUT TAB(1,ROW) NS
50 INPUT TAB(13,ROW) I
60 INPUT TAB(22,ROW) R
70 INPUT TAB(30,ROW) N
80 IF I < > N THEN PRINT TAB(35,ROW);
   INT((R/(I-N))*100)/100 ELSE
   PRINT TAB(35,ROW)"*"
90 LET ROW=ROW+1
100 GOTO 40

```

The average is worked out in Line 80. It looks rather complicated but this is only so the result is given to two decimal places. The straightforward average is runs divided by (innings minus 'not outs') but this would be calculated to 9 decimal places. This is much too accurate for our needs, and would also mess up the display as the numbers would run into each other.

One other complication is that if the batsman has as many 'not outs' as innings you will get an error report—'division by zero'—when it tries to work out the average. So Line 80 also checks whether I equals N, and if it does then it PRINTs out an asterisk in place of the average.

USING PUNCTUATION MARKS

As you have seen, TAB is used to position text and get INPUTs at any position on the screen.

But there are other ways of laying out or displaying information without using TAB. It is simply a matter of using the correct punctuation—either commas, semicolons or apostrophes. When you get the hang of these tricks you'll find that in a lot of cases, especially with columns of figures or words, the information can be positioned very easily with hardly any effort on your part and only the occasional TAB statement.

Try these next few lines to get a feel of how it works:

```

5 CLS
10 PRINT "01234567890123456789
   01234567890123456789"
20 PRINT 9;9;9
30 PRINT 9,9,9
40 PRINT 9'9'9

```

(Line 10 just numbers the columns along the top of the screen for reference.) Notice carefully how the punctuation affects where the numbers are PRINTed. The numbers separated by a semicolon are PRINTed next to each other. A comma separates the numbers into 'fields', each 10 columns wide. And an apostrophe puts each number on a new line.

Now try the same program with letters (that is, strings) instead of numbers. Add these lines and RUN the program again:

```

45 LET A$="A"
50 PRINT A$;A$;A$

```

```

60 PRINT A$,A$,A$
70 PRINT A$'A$'A$

```

Here you'll notice that the punctuation marks do the same thing but that strings are always PRINTed at the left of each 10-column field while numbers are PRINTed at the right.

This does actually make sense. If you wrote out a column of words you would normally line them up at the left, but you'd almost certainly line up a column of figures on the right so the units, tens, hundreds, and so on were all in line. And this is just what the computer does.

Type in and RUN this next program. See how the computer lines up the figures and then adds them up—just as you would if you were doing the sum on paper.

```

10 PRINT 14'67'245'42'811
20 PRINT '14 + 67 + 245 + 42 + 811

```

If you want several columns then use the comma. Here's a program to PRINT out the square, cube and fourth power of all numbers from 1 to 10.

```

10 CLS
20 PRINT TAB(4)"NUMBER" TAB(14)
   "SQUARE" TAB(26)"CUBE" TAB(31)
   "4TH POWER"
30 FOR J=1 TO 10
40 PRINT J,J*J,J*J*J,J*J*J*J
50 NEXT J

```

The numbers are PRINTed out at Line 40 and the commas make sure that each number is lined up in a separate field. Note that the headings to the columns do need to be positioned using TAB. If you used commas here, the headings would be lined up at the left of each field and would not be above the numbers.


The semicolon is also extremely useful. It is often used after a TAB statement so that what comes next is PRINTed without any spaces in between. The semicolon is not actually necessary if you are PRINTing out strings because, as you know by now, strings are always PRINTed to the left anyway. But it is essential when you're PRINTing numbers and you must put it in if you want the numbers PRINTed at the TAB position.

The semicolon is also useful at the end of a PRINT statement as it keeps the cursor there ready for the next PRINT item—which may come some time later. This last program PRINTs out the alphabet using a FOR . . . NEXT loop to step through the ASCII codes for the characters from A to Z:

```

10 CLS
20 FOR code=65 TO 90

```



1. The semicolon can be used with PRINT or INPUT to control how the information is displayed. Each item is put on the screen immediately following the one before, on the same line


```
30 PRINT CHR$(code);
40 NEXT code
```

Try it without the semicolon and you'll find that each letter is PRINTed on a separate line.

The punctuation marks and TAB offer such a variety of ways to display text on the screen that you're sure to find some combination that suits your particular program. It just takes a bit of practice to know exactly what each one can do.



The Sinclair machines have three statements to make things on the screen look neater: PRINT, PRINT AT and PRINT TAB.

PRINT, preceded by a line number and followed by nothing at all, is the simplest. It just PRINTs a blank line. A couple of examples are in the cricket program (see below).

PRINT AT you have probably met before but it deserves a slightly fuller explanation:

The Spectrum screen has 22 lines, numbered 0–21 from the top of the screen. It has 32 'columns', or character places, numbered 0–31 from the left of the screen. The line number is always given first. So these lines:

```
PRINT AT 0,0;"*"
PRINT AT 21, 0; "*"
PRINT AT 0, 31; "*"
PRINT AT 21, 31; "*"
```

will PRINT a star at each corner of the screen.

If you tell the computer to PRINT more than one character at a particular screen location, it will PRINT the first character where you tell it to, then just carry on in the same line.


```
PRINT AT 10, 12; "LENGTH"
```

... for example, puts the 'L' at 10, 12 and the other letters at 10, 13 to 10, 17 respectively.



This screen picture (this example was generated on the Spectrum) shows you how you can use a semicolon to join up your screen messages. Try a program like:

```
10 PRINT "message";
20 GOTO 10;
```



2. Apostrophes are not used on all machines. When they are available, they instruct the computer to put the next item on a new line, when it would not do so automatically

And if what you want displayed will not fit in the line, starting from the location you specify, it will simply overflow into the line below—sometimes changing the meaning in the process. Try this, for instance:

```
PRINT AT 5, 29; "TENANTS"
```

PRINT TAB

The PRINT TAB instruction on the Spectrum works in much the same way as PRINT AT. There are two differences, though:

1. You do not have to specify the line.
2. You cannot use PRINT TAB to overwrite—and therefore erase—any material on the screen. If you enter a new PRINT TAB at the old position the display drops down a line.

To make the difference clear in your mind, first try this:

```
PRINT AT 0, 15; "*"
PRINT AT 0, 15; "?"
```

Then clear the screen (CLS) and try this:

```
PRINT TAB 15; "*"
PRINT TAB 15; "?";
```

For reasons explained below, *be very careful with the punctuation.*

PRINT TAB has no great advantage over PRINT AT—until, that is, you have a lot of tabulated material to enter. Then it saves a lot of typing, and also having to remember line numbers as you go.

This little program, for example, allows you to enter a series of cricket scores and tabulates them neatly on the screen. It also works out the batsman's average. Enter all the lines in capitals on the ZX81:

```
10 PRINT TAB 14;"INNS";TAB 19;"RUNS";
    TAB 24;"N.O.";TAB 29;"AVG"
15 PRINT
20 INPUT "Name?";n$
30 PRINT TAB 0;n$;
40 INPUT "Innings?";i
50 INPUT "Runs?";r
60 INPUT "Not outs?";n
80 PRINT TAB 14;i;TAB 19;r;TAB 24;
    n;TAB 29;INT (r/(i - n))
90 PRINT
100 PAUSE 200
110 GOTO 20
```

The program looks neat on the screen partly because of the tabulation and partly because Line 90 PRINTs a blank line under each batsman's figures.

It is also easy to use. When you first RUN it, it will first ask you for each batsman's name, then his other details. When it asks for 'Runs?' there is no need to perform feats of mental agility to add his newest score to his previous total—you just let the computer do the work.

There are two snags in the program as it stands, though. The first is that the average calculated by the last bit of Line 80 is always

rounded *down* to a whole number. How to deal with such number complications is in a later article, but if this average is too coarse for you you can always enter a separate PRINT TAB line for the average and have it calculated to ten decimal places!

The second 'catch' is that should a batsman have as many 'not outs' as innings you will get an error report: Number too big. This is because this tail-ender's theoretical average is actually infinity!

To deal with the problem, delete the existing Line 80 and add these new lines:

```
70 IF i=n THEN PRINT TAB 14;i;TAB
    19;r;TAB 24;n;TAB 29;"*"
80 IF i < > n THEN PRINT TAB 14;i;TAB
    19;r;TAB 24;n;TAB 29;INT (r/(i-n))
```

To escape from the program without clearing the screen, you will need to press [CAPS SHIFT] and [BREAK] together ([BREAK] alone on the ZX81). Line 100 gives you time to do this before the computer demands another name.

A program to tabulate and calculate at the same time is also quite straightforward. Here is one simple example, the raw beginnings of a household accounts program:

```
10 PRINT TAB 0;"DATE";TAB 10;"ITEM";
    TAB 26;"COST"
15 PRINT : PRINT
20 LET total=0
30 FOR I=1 TO 8
40 INPUT "Date?";d$
50 INPUT "Item?";i$
60 INPUT "Cost?";c
70 PRINT TAB 0;d$;TAB 10;i$;TAB 26;c
75 PRINT
80 LET total = total + c
90 NEXT I
110 PRINT TAB 26;"- - - - -";
120 PRINT TAB 10;"Total";TAB 26;total
```

WATCHING THE DOTS

All your efforts to make a neat and tidy screen display will instantly be spoiled if you get the punctuation wrong, especially with a PRINT TAB statement.


Normally, two sets of semicolons are used after a PRINT TAB statement. To see why, try this tiny program:

```
10 FOR n= 1 TO 21
20 FOR t= 0 TO 24 STEP 6
30 PRINT TAB t; n;
40 NEXT t
50 NEXT n
```

When you have RUN it, change Line 30 to:

```
30 PRINT TAB t, n;
```

... and then compare it to this:



Another actual screen example shows the force of the apostrophe in a program line like:

```
10 PRINT "message" '
20 GOTO 10
```

```
30 PRINT TAB t; n
```

You can puzzle out, if you wish, why you get the results you do, but it is not important. What is important are these rules:

- 1 A semicolon means, 'close the next word (or whatever) up tight, without a space'.
- 2 A comma means, 'start the next word (or whatever) at the beginning of column 0 or column 15, whichever comes first'.
- 3 And a PRINT TAB statement needs *two* sets of semicolons, unless you want a terrible mess!



The easiest way to display a message on the screen is like this:

```
PRINT "HELLO"
```

'Hello' will appear at the far left of the screen on the next available line—if PRINT "HELLO" is on the last line of the screen the screen will scroll to make room for the message. You must be careful to type the inverted commas or else you will get an error report.

In this form all the programming clutter remains on the screen as well as the message. It would be better if the message appeared on a clear screen. Here is what you need to do:

```
10 CLS
20 PRINT "HELLO"
```

It's just like giving the computer a clean piece of paper to write on. 'Hello' now appears in the top left on an otherwise blank screen. Add this line and RUN the program:

```
30 PRINT "GOODBYE"
```

'Goodbye' will appear directly below 'Hello'. Suppose you want a gap between 'Hello' and 'Goodbye' to space out the display a little.

Add this line and see what happens:

```
25 PRINT
```

There will now be a blank line inserted between 'Hello' and 'Goodbye'.

You've now got some degree of control over how your messages are arranged vertically on the screen, but still no control over where the message will appear on each line. Suppose, for instance, you want to display 'Hello' towards the middle of the line. The computer has a command called PRINT TAB which allows you to place your message anywhere you like along the line.

When you add this line and RUN the program you'll see what effect PRINT TAB has:

```
40 PRINT TAB(15);"GOOD AFTERNOON"
```

The spaces on each line of the screen are numbered from 0 to 31 starting from the left. The message in Line 40, then, will start on space number 15. Be careful not to leave a space between TAB and the bracket—if you do (15) will be treated as a variable and TAB won't work. Also be sure to include the semicolon.

TAB is short for 'tabulate'. This program shows you how useful TAB can be:

```
10 CLS
20 PRINT"NUMBER";TAB(7);"CUBE";
    TAB(13);"SQR";TAB(18);"SQR-RT";
    TAB(26);"RECIP"
30 FOR J=1 TO 12
40 PRINT TAB(1);J;TAB(6);J*J*J;
    TAB(12);J*J;TAB(17);INT(SQR(J)*
    1000)/1000;TAB(25);INT(1000/J)/
    1000
50 NEXT J
```

When you RUN the program you'll see a table of numbers being built up. Every time the program goes through the loop another line of figures is displayed on the screen. The use of PRINT TAB allows you to place each figure in the right column.

If you are puzzled by the multiplying and dividing by 1000 in Line 40 the explanation is that this yields a figure correct to three decimal places.

Suppose you want to PRINT at a specific place on the screen—not just on the next available line. In this case PRINT TAB isn't good enough. Instead you must use PRINT@. Try this program:

```
10 CLS
20 PRINT@ 463,"LOW"
30 PRINT@ 71,"HIGH"
40 PRINT@ 257,"LEFT"
50 PRINT@ 282,"RIGHT"
```

You'll see that the words appear in the appropriate positions on the screen.

The numbers after PRINT@ refer to screen locations, which on the Dragon and Tandy are numbered from left to right starting at the top of the screen.

Here is a program which uses both PRINT@ and PRINT TAB:

```

10 CLS
20 PRINT "NAME";TAB(7);"INNS";TAB(13);
  "N-O'S";TAB(19);"RUNS";TAB(26);"AV"
30 PA = 32
40 PRINT@448,"":PRINT@448,"NAME□";:
  INPUT$
50 PRINT@448,"":PRINT@448,
  "INNINGS□";:INPUT IN
60 PRINT@448,"":PRINT@448,"NOT-
  OUTS□";:INPUT NO
70 PRINT@448,"":PRINT@448,"RUNS□";:
  INPUT RU
80 PRINT@PA,NS;TAB(7);IN;TAB(12);NO;
  TAB(18);RU;TAB(24);
90 IF NO = IN THEN PRINT"□*" ELSE
  PRINT INT(100*RU/(IN - NO))/100
100 PA = PA + 32
110 IF PA < 448 THEN GOTO 40 ELSE END

```

When you RUN the program Line 20 displays the column headings at the top of the screen. Lines 40 to 70 clear a line of the screen and ask for information.

Line 80 tabulates the information you have just given the machine, and Line 90 calculates the average. If the average is going to be infinity—caused by the machine trying to divide by zero—an asterisk is displayed.

The variable PA allows the machine to check that the screen hasn't been filled. When the table has filled the screen the program stops.

PUNCTUATION

Punctuation is very important when you are displaying information on the screen. Commas and semicolons control the position of the cursor—the flashing black square under which the characters appear. Type in:

```

10 CLS
20 PRINT "THE CURSOR WILL RETURN"
30 PRINT "WITHOUT A SEMICOLON"
40 PRINT "BUT WITH A SEMICOLON□";
50 PRINT "THIS HAPPENS"

```

When you include a semicolon in a program line—but *outside* inverted commas—the cursor doesn't return to the beginning of the next line. The cursor stays exactly where it was when it finished PRINTing. The next PRINT statement continues directly after the last one.

Add these lines to the program and RUN it:

```

60 PRINT "BUT WHAT ABOUT",
70 PRINT "USING",
80 PRINT "COMMAS",
90 PRINT "?"

```

The computer's screen is divided into two halves. When you put a comma at the end of a PRINT line the cursor will jump to the next half of the screen. Commas can, therefore, provide you with a rudimentary form of tabulation.



Cursor controls, punctuation, POKES, PRINTing functions and spaces can all be used—singly or together—to smarten up your Commodore screen displays.

PRINTING FUNCTIONS

Perhaps the most commonly used of the cursor positioning instructions is the TAB function. This behaves, on screen, like the tab facility of a typewriter, positioning the cursor so many character spaces, or columns, from the left-hand edge of the screen text area. It takes the form:

```
10 PRINT TAB(10) "HELLO"
```

You can place a semicolon after the brackets, but this isn't necessary. There must be *no* space between TAB and its bracketed *argument*, the value of which is the starting point of the PRINTed string.

Multiple TABs may be used to position several comments on each screen line, or row:


```
10 A$ = "GOOD" : B$ = "MORNING"
20 PRINT TAB(5)A$ TAB(10)B$
```

The result is that G is printed in column 5 and M in column 10—the argument value always starts at the left-hand edge of the text area.

Now change Line 20 as follows:

```
20 PRINT TAB(5)A$ SPC(5)B$
```

Note that how, this time, the function SPC creates 5 spaces between the two words,



Again from the Spectrum screen, the comma will PRINT in columns with a program like:

```
10 PRINT "message",
20 GOTO 10
```

Acorn and Commodore computers have more columns available.

taking the end of the previous word as the starting point. Try changing the argument to 115: note how the spacing continues over several lines. Values over 255 cannot be used.

Several SPC functions may be included in the same PRINT statement, but remember that your screen is restricted to 25 lines of 40 columns, and the total character spaces cannot exceed this.

Both TAB and SPC can make use of variables in their argument, as in this program. On the Vic 20, change the 10 in Lines 10 and 40 to a 40.

```

5 PRINT"□"
10 FORL = 1 TO 10
20 PRINT TAB(L)"□"
  SPC(L + 1)"."SPC(L + 2)
  "□"
30 NEXT : R = L
40 FORL = 10 TO 1 STEP - 1
50 PRINT TAB(L)"□" SPC(R + 1)
  "."SPC(R + 2)"□"
60 NEXT

```

This PRINTs a simple roadway perspective. It is an example of some of the unusual uses you can find for functions you'd normally use for 'information' screens, such as menus.

Not often discussed but nevertheless a function belonging to the same group is POS, which gives the value of the column (0 to 39) where the *next* PRINT position begins. This program PRINTs a random roadway. On the Vic change the 20 in Line 10 to 6, and the 25 in Line 30 to 12:

```

10 PRINT "□": LET A = 1: LET Z = 20
20 LET Z = Z + A: PRINT TAB(Z)
  "↑□□□↑";
30 IF POS(0) > 25 + RND(1)*9 OR POS(0)
  < 5 + RND(1)*9 THEN A = - A
40 PRINT:GOTO 20

```

The value following POS, the argument, is only a dummy value in this case.

USING SPACES

Simple screen formatting tasks can often be carried out by incorporating spaces within PRINT statements, and this often uses less memory for small spaces.

```
10 PRINT "□□□HELLO THERE"
```

However, it is quite in order to include really long space strings where these are themselves used for formatting purposes in conjunction with 'string slicing' (covered in a later article).

One of the less obvious advantages of using spaces in these ways is that they do PRINT as blanks, thus overwriting any screen in formation which may be present. TAB, SPC and

MULTIPLE LETTERS MADE EASY

Writing large numbers of similar letters—job applications, say—can be a long and tedious chore if you have to do it by hand.

The program with this article aims to take the drudgery out of such jobs. It is a simplified kind of text editor which allows you to enter one master letter and then produce repetition copies, each one slightly different if necessary. Some keyboards make this much easier than others, and note that there is no program for the ZX81 because of the limitations of its key pad. 'Real' keyboards are available as accessories for the Spectrum.

Unlike a proper word-processing program, which would take hours to type in by hand, you cannot (except on Acorn machines) see exactly on your TV screen how the letter will look. But it doesn't matter: the program itself automatically avoids word 'splits' at the ends of lines. It also automatically puts a line of space between paragraphs—neater and more professional than indenting them.

For important letters such as job applications, a good printer (and ribbon!) is obviously necessary. These are expensive, but if you do not yet own a printer, perhaps you could borrow the use of a friend's, or one at your local computer club. Printers are covered in detail in a later article.

ENTERING THE PROGRAM

Start by typing in the main program, without any DATA statements. Then SAVE it on tape or disk so you can use it for any letter.

Before you print out your first letter you may need to make one small adjustment, depending on the printer you propose to use.

First find out how many characters in a line your printer will accept. Then adjust your program both to suit this line length, and to allow for a suitable margin on either side. In all the programs, TL represents the total available width, and LL the width of your typing.

You can of course change these to suit yourself—if necessary, by making a last-minute adjustment to the program just before you start printing out your letters.

Exactly how you make this adjustment varies from one computer to another:



Amend Line 30 in the program so that TL is the available printer width, and LL the width you want your letter.

If you want to display your letter on the TV screen so that you can get an idea of what your letter will look like when printed, temporarily delete Line 70. But reinstate it before you go to the printer.

Do you make mistakes when you're writing important letters? If so, here's a program which lets you edit a master letter, then produce copies to send to different people.



Amend Line 10 in the program so that PL is the available printer width, and LL the width you want your letter. (In the example given, the widths are those available on the Sinclair ZX printer although for letter quality printing, this is not really suitable.)



Amend Line 20 so that TL is the available printer width, and Line 30 so that LL is the width you want your letter.

As it stands, Line 40 gives you the chance to see where the word 'turns' are on the TV screen. Before your letter is printed, change P=0 to P=2.





Amend Line 10 so that TL is the available printer width, and also change it so that LL is the width you want your letter.

Note: The Commodore program has been tested on a Commodore printer only.

ENTERING YOUR FIRST LETTER

Your letter consists of a series of DATA statements. Enter them as shown in the sample on page 128 (it is for the Dragon, but all the programs and entries you need on



■	HOW TO ENTER THE PROGRAM
■	ADJUSTING THE PROGRAM TO SUIT YOUR PRINTER
■	LAYING OUT LETTERS WITH

■	ENTERING A LETTER
■	ENTERING AND AMENDING YOUR MASTER LETTER
■	PRINTING OUT A HARD COPY

the other computers work in the same way). The DATA must start in Line 1000, and must start with your address—it would be an unusual letter that didn't!

The first thing the computer will do is to search through the lines containing your address, find the longest line, and arrange the whole address neatly down the right-hand side of the paper.

After that, the computer will make each line start flush left, the modern way of setting out a business letter.

Note that each line of your letter must start with quotation marks, as with most DATA statements. Here's a list of what the other symbols mean:

- The hash mark, #, means 'this line is part of my address. Set it to the right of the page'.
- The dollar sign, \$, means 'start a new paragraph, first leaving a full line of spaces above it'.
- The ampersand, &, means 'start a new line, flush left, but do not leave a space above it'. (This is useful for the address of the person to whom you are writing.)
- The multiplication sign, *, means 'centre this line (only)'.

If you want extra line spaces—for example, between 'Yours faithfully' and your

signature—all you have to do is insert some extra dollar signs, each enclosed in inverted commas, as shown.

When you reach the end of your letter, two computers require an extra line:



Enter an extra line number, followed by DATA \$

Enter an extra line number, followed by DATA

AMENDING YOUR LETTER

Since it is all in BASIC, you can if you wish SAVE your first (or draft) letter as part of the master program.

In this case, the first step in amending a letter is to LOAD the program plus DATA. Then you can alter any line simply by editing it in the usual way.

If you forget to SAVE your first letter, you can still produce an amended one—but you will have to type in the entire DATA instead of just a few amendments.

PRINTING THE LETTER

To print out your letter, you merely RUN the program, which turns the printer on automatically and then turns it off again after printing.

If you try to RUN the program without having a printer connected, and without remembering to delete Line 70, notice that the beginning of the letter will appear on the screen then the computer will wait for it to be printed. To regain control, you must break into the program, delete Line 70 and RUN again. This time you'll be able to see the whole of the letter on the screen.

The type of paper on which you print depends on the printer you use. Some printers give you a choice of either single sheets, which are best for formal correspondence, or folded paper. If you use folded paper, turn the paper forwards so that your letter begins on a whole page. Once the edges have been torn along the perforations, the page should be neat, but you could trim it with a sharp knife if necessary.



If you have headed paper, then you will not want to print your address again. You must still start the DATA at Line 1000 but in this case it would start with the address of the person you are writing to, and you'll not need to use the hash '#' sign.



10 ON ERROR GOTO 480

20 MODE0

30 LL=60:TL=80




```

40 C$ = ""
50 LL = LL + 1
60 D$ = STRING$( (TL - LL) / 2, "□")
70 VDU2
80 PRINT D$;
90 READ A$
100 A$ = A$ + "□"
110 IF ASC(A$) = 35 THEN
    PROCADDR
120 IF ASC(A$) = 36 THEN PRINT "D$;";
    A$ = RIGHT$(A$, LEN(A$) - 1); LP = 0
130 IF ASC(A$) = 38 THEN PRINT "D$;";
    A$ = RIGHT$(A$, LEN(A$) - 1); LP = 0
140 IF ASC(A$) = 42 THEN PROCENT:
    GOTO 90
150 C$ = ""
160 FOR T = 1 TO LEN(A$)
170 B$ = MID$(A$, T, 1)
180 C$ = C$ + B$
190 IF LEN(C$) > LL THEN VDU3:
    PRINT "FORMAT ERROR—WORD
    TOO LONG": END
200 LP = LP + 1
210 IF LP <= LL THEN 240
220 PRINT "D$;"; IF ASC(C$) = 32 THEN
    C$ = RIGHT$(C$, LEN(C$) - 1)
230 LP = LEN(C$)
240 IF B$ = "□" AND LP <= LL THEN
    PRINT C$; C$ = ""
250 NEXT
260 GOTO 90
270 DEF PROCENT
280 A$ = RIGHT$(A$, LEN(A$) - 1)
290 IF LEN(A$) > LL THEN VDU3:
    PRINT "LINE TOO LONG—CANNOT
    CENTRE IT": END
300 PRINT "D$;"; STRING$(
    ((LL - LEN(A$)) / 2, "□"); A$; "D$;";
310 LP = 0
320 ENDPROC
330 DEF PROCADDR
340 X = LEN(A$)
350 RESTORE
360 READ A$: IF ASC(A$) < > 35 THEN
    390
370 IF LEN(A$) > X THEN X = LEN(A$)
380 GOTO 360
390 IF X > LL THEN VDU3: PRINT
    "FORMAT ERROR—ADDRESS TOO
    LONG": END
400 E$ = STRING$( (LL - X), "□")
410 RESTORE
420 READ A$
430 REPEAT
440 A$ = RIGHT$(A$, LEN(A$) - 1)
450 PRINT E$ A$ D$;
460 READ A$
470 UNTIL ASC(A$) < > 35
475 ENDPROC
480 PRINT: VDU3: MODE6: REPORT:
    PRINT: END
    
```




```

5
10 LET LL = 32: LET PL = 32
15 LET LL = LL + 1: LET T = (PL - LL) / 2
20 LET D = 0
30 READ A$: LET L = LEN A$
40 LET C = 0
50 IF C = L THEN GOTO 30
60 LET C = C + 1: LET D = D + 1: IF C > 1
    THEN GOTO 100
70 IF A$(C) = "#" THEN GOTO 500
80 IF A$(C) = "*" THEN GOTO 700
85 IF A$(C) = "&" THEN GOTO 850
90 IF A$(C) = "$" THEN LPRINT
    CHR$ 13; CHR$ 13;: LET D = 0:
    GOTO 900
    
```

```

95 LET A$ = "□" + A$: LET L = L + 1
100 IF A$(C) = "□" THEN GOTO 800
110 LPRINT A$(C);
115 IF D > LL THEN LET D = 0
120 GOTO 50
500 LET NL = 0: LET TA = LL: LET BE = 0
510 LET LE = LEN A$ - 1: IF LE > LL
    THEN PRINT FLASH 1; "FORMAT ERROR—
    ADDRESS TOO LONG": STOP
520 IF LE > BE THEN LET BE = LE
530 LET NL = NL + 1: READ A$: IF
    A$(1) = "#" THEN GO TO 510
540 RESTORE 1000
550 LET TR = T + LL - BE: FOR G = 1 TO NL:
    FOR H = 1 TO TR: LPRINT "□";: NEXT H:
    
```

```

READ A$: LPRINT A$(2 TO ): NEXT G
560 GOTO 30
700 LET TA = (LL - L)/2 + T: IF TA < T
THEN LPRINT CHR$ 13: PRINT FLASH
1;"FORMAT ERROR—CANNOT CENTRE":
STOP
710 LPRINT CHR$ 13;: FOR N = 1 TO
TA: LPRINT "□";: NEXT N: LPRINT
A$(2 TO L): GOTO 20
800 LET SL = LL - D - 1: LET CC = C + 1:
LET X = 1
810 IF CC = L THEN GOTO 825
820 IF A$(CC) < > "□" THEN LET
CC = CC + 1: LET X = X + 1: GOTO 810
825 IF X > = LL THEN LPRINT CHR$ 13:

```

```

PRINT FLASH 1;"FORMAT ERROR—WORD
TOO LONG":STOP
830 IF SL > = X THEN GOTO 110
850 LPRINT CHR$ 13;: LET D = 0
900 FOR B = 1 TO T: LPRINT "□";:
NEXT B: GOTO 50

```



```

10 CLEAR 200
20 TL = 80
30 LL = 56
40 P = 0
50 SP = (TL - LL)/2:HW = TL/2
60 PRINT # - P,CHR$(13)
70 IF P = 2 THEN SP$ = STRING$(SP,"□")
ELSE SP = 0:HW = 16
200 READA$:A$ = A$ + "□"
210 IF A$ = "" GOTO 290
220 O$ = LEFT$(A$,1)
230 IF O$ = "#" THEN ML = 0:
GOSUB 400:GOTO 290
240 IF O$ = "$" THEN RL = 0:PRINT # - P,
CHR$(13);CHR$(13);SP$;:ST = 2:
GOTO 280
250 IF O$ = "*" GOSUB 800:
RL = HW - LEN(A$)/2:GOTO 280
260 IF O$ = "&" THEN RL = 0:PRINT # - P,
CHR$(13);SP$;:ST = 2:GOTO 280
270 ST = 1
280 GOSUB 600
290 GOTO 200
400 IF LEN(A$) > ML THEN ML =
LEN(A$)
410 N = N + 1:READA$:IF LEFT$(A$,1) =
"#" GOTO 400
420 IF ML > LL THEN CLS:PRINT" format
error□ . . . ADDRESS IS TOO LONG
FOR THE SET LINE-LENGTH!":END
430 RESTORE:FOR J = 0 TO N - 1:
READA$:PRINT # - P,STRING$
(LL - ML,"□");SP$;:PRINT # - P,
MID$(A$,2);CHR$(13);
440 NEXT:RETURN
600 WL = 0
610 IF ST + WL > LEN(A$) THEN 630
620 IF MID$(A$,ST + WL,1) < > "□"
THEN WL = WL + 1:GOTO 610
630 IF WL > LL THEN CLS:PRINT"format
error . . .":A$:PRINT"□ . . . CONTAINS
A WORD LONGER THAN THE LINE-
LENGTH!":END
640 IF RL + WL - 1 > LL THEN PRINT
# - P,CHR$(13);SP$;:RL = 0
650 WL = WL + 1
660 PRINT # - P,MID$(A$,ST,WL);:
RL = RL + LEN(MID$(A$,ST,WL))
670 ST = ST + WL:IF ST < LEN(A$) + 1
GOTO 600
680 RETURN
800 IF LEN(A$) > LL THEN CLS:
PRINT" format error . . . CANNOT

```

```

CENTRE",A$:END
810 PRINT # - P,CHR$(13);
820 IF HW > LEN(A$)/2 THEN PRINT
# - P,STRING$(HW - LEN(A$)/2,"□");
830 ST = 2:RETURN

```



What should a job application look like?

The layout and presentation of a job application is extremely important as it is the first impression the company has of you—a jumbled up or cramped letter will leave a bad impression no matter how well qualified you are. Letter-writing styles change over the years but here are the latest rules.

Your address goes at the top right of the paper, but not too near the top. You can leave out all punctuation marks even after abbreviations such as Rd or St for Road and Street and you should start each line directly below the one above, without staggering them.

The date goes under your address after a line space if you like.

Next comes the name and address of the person you're writing to. This goes on the left-hand side of the page, either level with or below your address. The name and address should be written out in full as it will be your record of who you sent the letter to—assuming you remember to keep a copy or to SAVE it on tape, that is!

If the job advertisement gave a reference number then you should put 'Your ref' followed by the number above their address.

Now comes the letter itself. If you're applying for a particular post then write this in the centre of the line below 'Dear Sir' or whatever, and leave a space above and below so the person who has to process the application can see at a glance what the letter is about.

The main body of the letter looks neatest if the paragraphs all start at the left with a space between each one.


Finally, 'Yours faithfully' or 'Yours sincerely' can go at the left, right or centre of the page with your name lined up underneath leaving enough space for your signature. By the way, the rule is use 'faithfully' with 'Dear Sir' and 'sincerely' with a persons name

A typical letter entered as a series of DATA statements on the Dragon—and the letter as it appears from the printer, ready to sign and post off





```

10 CLR:LL=60:TL=80:QQ=(TL-LL)/2:
  DIM A$(20):PRINT"□"
15 FOR Z=1 TO LL+QQ:SS=$$+"□":
  NEXT Z
20 OPEN 4,4:CMD 4:LE=0:GOTO 800
100 READ K$
102 K$=K$+"□":IF LEFT$(K$,1)="*"
  THEN LE=0:GOTO 600
110 IF K$="□" THEN PRINT #4,
  CHR$(13):CLOSE 4:END
115 IF LEFT$(K$,1)="&" THEN LE=
  0:PRINT #4,CHR$(13):GOTO 400
120 IF LEFT$(K$,1)="$" THEN
  PRINT #4,CHR$(13):LE=0:GOTO 400
130 GOTO 452
400 K$=RIGHT$(K$,LEN(K$)-1)
450 IF LEN(K$)<1 THEN 100
452 IF LE=0 THEN PRINT #4,
  LEFT$(SS,QQ);
455 FOR L=1 TO LEN(K$)
460 IF MID$(K$,L,1)="□" THEN
  KKS=LEFT$(K$,L):K$=RIGHT$(K$,
  LEN(K$)-L):GOTO 480
470 NEXT:GOTO 100
480 GOSUB 500:GOTO 450
500 IF LEN(KK$)>LL+1 THEN 950
505 IF LE+LEN(KK$)>LL THEN
  PRINT #4,CHR$(13):LEFT$(SS,QQ):
  LE=0
510 LE=LE+LEN(KK$):PRINT #4,
  KK$:RETURN
550 RETURN
600 PRINT #4,CHR$(13):LEFT$(SS,
  QQ):IF LEN(K$)>LL THEN 960
610 PRINT #4,LEFT$(SS,(LL*.5)-
  (LEN(K$)-1)*.5):RIGHT$(K$,
  LEN(K$)-1):GOTO 100
800 READ X$:P=P+1:IF LEFT$(X$,1)
  <>"#" THEN 900
810 A$(P)=RIGHT$(X$,LEN(X$)-1):
  IF LEN(A$(P))>HL THEN
  HL=LEN(A$(P))
815 IF HL>LL THEN 950
820 GOTO 800
900 IF P=1 THEN RESTORE:GOTO 100
910 P=P-1:FOR Z=1 TO P
920 PRINT #4,LEFT$(SS,QQ):
  LEFT$(SS,(LL-HL))A$(Z):NEXT Z:
  K$=X$:GOTO 102
950 PRINT #4,"□":PRINT "FORMAT
  ERROR—WORD TOO LONG":CLOSE 4:
  END
960 PRINT #4,CHR$(13):PRINT
  "FORMAT ERROR—CANNOT CENTRE
  LINE":CLOSE 4:END
  
```



COMING IN ISSUE 5 ...

- ▣ Arcade-type games are much less exciting if the 'enemy' doesn't shoot back at you. Learn how to create **DEADLY ENEMIES**, and how to protect yourself from them with shields.
- ▣ Put your computer to work again, with a new program that stores and calculates your **FINANCIAL ACCOUNTS** for personal expenditure.
- ▣ Move one step nearer to your own machine code programs by getting to grips with **HEXADECIMAL ARITHMETIC** – the base-16 system the computer itself uses.
- ▣ Tell your computer what to do, using the **BASIC** commands **INKEY\$**, or **GET** and **GET\$**. There's even a graphics program that gives you fingertip control through the keyboard.
- ▣ Learn about **ARRAYS**, the powerful **BASIC** tools that lets a programmer store information until it's needed.



ASK YOUR NEWSAGENT FOR INPUT